
TEXgraph, exemples de graphiques

P.Fradin

26 avril 2011

F F′

M

Figure 1

\begin{texgraph}[name=figure1, file]
Graph image = [
view(-4, 4, -1, 4), Marges(0, 0, 0, 0), size(8, 1),
Axes(0, 0), F :=-3, F' :=3, A :=2+2.5*i,
Ligne([F, A, F'], 0),
Arrows :=1, Width :=8, Ligne([0, 1, jump, 0, i], 0), Arrows :=0,
DotScale :=1.15, DotStyle :=cross,
LabelDot(F, "$F$", "S", 1), LabelDot(F', "$F'$", "S", 1), LabelDot(A, "$M$", "N", 1)
] ;
\end{texgraph}

Figure 2

\begin{texgraph}[name=figure2, file]
Include "draw2d.mod" ;
Mac

legende = if %1=1 then "Point ordinaire"
elif %1=2 then "Point d'inflexion"
elif %1=3 then "Rebroussement 1ère espèce"
else "Rebroussement 2ème espèce"
fi ;

Graph objet1 =
[size(12), Width :=6,
{points de contrôle bezier1}
A :=(1+i)*0.75, B :=0.75, C :=0,
{points de controle bezier2}
A' :=[-1.25+i*0.5, -0.75-i*0.5, 1.25-0.5*i, 0.5+i], B' :=[-0.75, -0.75, 0.75, 0.75],
{vecteurs de tanslation}
u :=[-3+2.5*i, 3+2.5*i, -3-2.5*i, 3-2.5*i],
for k from 1 to 4 do

SaveAttr(),
SetMatrix([u[k], 1.5, i*1.5+1]), {le dessin se fait dans [-1, 1]x[-1, 1]}
{axes}
Ligne([-1, 1, jump, -i, i], 0),
{courbes bezier1 et bezier2}
Width :=8, Color :=blue,
draw("path", [A'[k], (B'[k]+2*A'[k])/3, (C+5*B'[k])/6, C, bezier,

(C+5*B)/6, (B+2*A)/3, A, bezier],
[marker :=[0.25, Carrow, 0.75, Carrow]]),

{vecteurs}
Color :=red, Arrows :=1, Ligne([0, 0.75, jump, 0, 0.75*i], 0), Arrows :=0,
{noms vecteurs}
Color :=black, LabelDot(1, "$f^{(p)}(t_0)$", "E", 0, 0.35),
LabelDot(0.75*i, "$f^{(q)}(t_0)$", "O", 0, 0.35),
{légende}
LabelDot(-i+1, @legende(k), "S", 0),
RestoreAttr()

od] ;
\end{texgraph}

Figure 3

\begin{texgraph}[name=figure3, file]
Include "variations.mod" ;
Mac

parO = [
{parO(lig, col, lig, col)}
u :=coinNO(%1, %2)-0.125*i, $v :=coinSO(%3, %4)+0.125*i,
mil :=(u+v)/2, ep :=(%3-%1)/8,
Bezier(u, mil-ep, v) ] ;

parF = [
{parF(lig, col, lig, col)}
u :=coinNE(%1, %2)-0.125*i, $v :=coinSE(%3, %4)+0.125*i,
mil :=(u+v)/2, ep :=(%3-%1)/8,
Bezier(u, mil+ep, v)] ;

Graph objet1 = [
dimTab(8, 11), defLargeCol(0, 1.5, [8, 11], 2), defHautLig(0, 1.5, 6, 2), size(12.5),
Cadre(6, 2, 6, 4, [LineStyle :=noline, FillStyle :=full, FillColor :=pink]),
Cadre(1, 8, 3, 8, [LineStyle :=noline, FillStyle :=full, FillColor :=pink]),
Cadre(6, 8, 6, 8, [LineStyle :=noline, FillStyle :=full, FillColor :=pink]),
valeursLig(1, +4, "b_{11}", "\cdots", "b_{1j}", "\cdots", "b_{1q}"),
valeursLig(2, +3, "B=", "\vdots", +1, "\vdots", +1, "\vdots"),
valeursLig(3, +4, "b_{p1}", "\cdots", "b_{pj}", "\cdots", "b_{pq}"),
valeursLig(4, "a_{11}", "\cdots", "a_{1p}", +1, "\bullet", "\cdots", +1, "\cdots", "\bullet"),
valeursLig(5, "\vdots", +1, "\vdots", +1, "\vdots", +3, "\vdots"),
valeursLig(6, -1, "A=", "a_{i1}", "\cdots", "a_{ip}", +1, "\bullet", "\cdots",

"\sum\limits_{k=1}^p a_{ik}b_{kj}", "\cdots", "\bullet", "=AB"),
valeursLig(7, "\vdots", +1, "\vdots", +1, "\vdots", +3, "\vdots", ),
valeursLig(8, "a_{n1}", "\cdots", "a_{np}", +1, "\bullet", "\cdots", +1, "\cdots", "\bullet", ),
Width :=8, Color :=darkblue,
LabelDot( pos(3, 4), "$\times$", "N"), LabelDot(coinNE(5, 8), "$=$", "O"),
Cadre(6, 2, 6, 4), Cadre(1, 8, 3, 8), Cadre(6, 8, 6, 8),
Arrows :=1, Bezier((coinNE(6, 3)+coinNO(6, 3))/2, pos(2, 3), (coinSO(2, 8)+coinNO(2,

8))/2),
Ligne( [(coinSO(3, 8)+coinSE(3, 8))/2, (coinNO(6, 8)+coinNE(6, 8))/2], 0),
Color :=black, Arrows :=0, Width :=6,
parO(4, 2, 8, 2), parO(1, 6, 3, 6), parO(4, 6, 8, 6),
parF(4, 4, 8, 4), parF(1, 10, 3, 10), parF(4, 10, 8, 10)
] ;

\end{texgraph}

A B

A∩B

Figure 4

\begin{texgraph}[name=figure4, file]
Graph image = [
view(-3, 3, -3, 3), Marges(0, 0, 0, 0), Width :=8, size(6),
A :=setB("A", -1, [scale :=0.5, rotation :=15]),
B :=setB("B", 1, [scale :=0.5, rotation :=-15]),
C :=capB(A, B),
FillColor :=lightgray, FillStyle :=full,
Color :=crimson, Bezier(C), LabelStyle :=framed, Label(-2.5*i, "$A\cap B$")
] ;
\end{texgraph}

Figure 5

\begin{texgraph}[name=figure5, file]
Include "variations.mod" ;
Graph image = [
defLargeCol(1, 1.5),
InitLig1("x", "-1", "0", "+1"),
InitCol1(+1, "\arccos\ ", +1),
variationsLig(2, "\pi",

-1,
["\frac{\pi}2", labelstyle :=framed, arrows :=0],
-1,
"0"

),
Width :=8, traitH(1), traitV(1)
] ;
\end{texgraph}

Figure 6

\begin{texgraph}[name=figure6, file]
Graph image = [
view(-5.5, 5.5, -3.5, 3.5), Marges(0, 0, 0, 0), size(12, 1),
P :=[0, 0, vecK],
SaveAttr(), FillStyle :=full, FillColor :=steelblue,
DrawPlan(P, vecI, 8, 8, 5),
RestoreAttr(),
Arrows :=1, dep :=[0, 0], Width :=8,
Arc3D(vecI, [0, 0], vecJ, 3, 1), Width :=12,
Ligne3D([dep, dep+2.5*vecI, jump, dep, dep+2.5*vecJ, jump, dep, dep+3*vecK], 0),
Width :=4, Arrows :=0,
angleD(Proj3D(vecI), 0, Proj3D(vecJ), 0.4),
angleD(Proj3D(vecK), 0, Proj3D(-vecJ), 0.4),
LabelDot(Proj3D(1.5*vecI), "$\vec{u}$", "O"),
LabelDot(Proj3D(2*vecJ), "$\vec{v}$", "N"),
LabelDot(Proj3D(2*vecK), "$\vec{w}$", "E"),
LabelDot(Proj3D(2*[1+i, 0]), "$+$", "S"),
LabelDot(Proj3D(4*[1+i, 0]), "$\mathcal{P}$", "SE")
] ;
\end{texgraph}

x 1 +∞

argch

0

+∞

Figure 7

\begin{texgraph}[name=figure7, file]
Include "variations.mod" ;
Graph image = [
InitLig1("x", "1", "+\infty"),
InitCol1(+1, "\argch\,\,", +1),
variationsLig(4, "0",

+2,
"+\infty"

),
Width :=8, traitH(1), traitV(1)
] ;
\end{texgraph}

Figure 8

\begin{texgraph}[name=figure8, file]
Graph image = [
view(-1.5, 3.5, -1.5, 3.5), Marges(0, 0, 0, 0), size(8, 0),
Arrows :=1, Axes(0, 0), Arrows :=0,
Droite(1, -1, 0),
LineStyle :=dashed, Ligne([-1, -1+i*pi, i*pi, jump, pi, pi-i, -i], 0),
LineStyle :=solid, tMin :=0, tMax :=pi,
Width :=8, Cartesienne(cos(x)),
LabelDot(pi-i, "$\mathcal{C}_{\cos}$", "S"),
DotStyle :=cross, DotScale :=1.5,
LabelAxe(x, -1, "$-1$", 1, 1), LabelAxe(x, 1, "$1$", 1, 1),
LabelAxe(x, pi/2, "$\frac{\pi}2$", 2, 1), LabelAxe(x, pi, "$\pi$", 2, 1),
LabelAxe(y, i, "$1$", 2, 1), LabelAxe(y, -i, "$-1$", 2, 1),
LabelAxe(y, i*pi/2, "$\frac{\pi}2$", 1, 1), LabelAxe(y, i*pi, "$\pi$", 1, 1),
tMin :=-1, tMax :=0.9999999,
Width :=8, Color :=red, Cartesienne(arccos(x)),
LabelDot(-1+pi*i, "$\mathcal{C}_{\arccos}$", "N")
] ;
\end{texgraph}

Figure 9

\begin{texgraph}[name=figure9, file]
Graph image = [
view(-4, 4, -4, 4), Marges(0, 0, 0, 0), size(8, 0),
Arrows :=1, Axes(0, 0), Arrows :=0,
Droite(1, -1, 0),
LineStyle :=dashed, Droite(1, 0, pi/2), Droite(1, 0, -pi/2),
Droite(0, 1, pi/2), Droite(0, 1, -pi/2),
LineStyle :=solid, tMin :=-pi/2, tMax :=pi/2,
Width :=8, Cartesienne(tan(x)),
LabelDot(1.5+i*3.75, "$\mathcal{C}_{\tan}$", "O"),
DotStyle :=cross, DotScale :=1.5,
LabelAxe(x, pi/2, "$\frac{\pi}2$", [1+i, 0.3], 1),
LabelAxe(x, -pi/2, "$-\frac{\pi}2$", [1-i, 0.3], 1),
LabelAxe(y, i*pi/2, "$\frac{\pi}2$", [1+i, 0.3], 1),
LabelAxe(y, -i*pi/2, "$-\frac{\pi}2$", [1-i, 0.3], 1),
tMin :=-5, tMax :=5,
Width :=8, Color :=red, Cartesienne(arctan(x)),
LabelDot(2+arctan(2)*i, "$\mathcal{C}_{\arctan}$", "SE")
] ;
\end{texgraph}

M1

M2

M3

M4

M5

M6

M0

1−1

1

−1

Figure 10

\begin{texgraph}[name=figure10, file]
Graph image = [
Marges(0, 0, 0, 0), view(-1.25, 1.5, -1.25, 1.25), size(6),
Width :=4, Arrows :=1, xylabelpos :=bottom+right, Axes(0, 0), Arrows :=0,
Width :=6, Color :=forestgreen, Cercle(0, 1), Color :=blue,
P :=for k from 0 to 6 do exp(2*i*k*pi/7) od, Ligne(P, 1), k :=1,
Width :=8, DotStyle :=bigdot, Color :=red, Point(P), Color :=black,
for z in P[2, 0] do Label(1.15*z, ["$M_", k, "$"]), Inc(k, 1) od,
LabelDot(1.15, "$M_0$", "N", 0),
Color :=black, DotStyle :=cross, DotScale :=1.5,
LabelAxe(x, 1, "$1$", 1+i, 1), LabelAxe(x, -1, "$-1$", 1-i, 1),
LabelAxe(y, i, "$1$", 1+i, 1), LabelAxe(y, -i, "$-1$", 1-i, 1)
] ;
\end{texgraph}

Figure 11

\begin{texgraph}[name=figure11, file]
Include "variations.mod" ;
Graph image = [
InitLig1("t", "0", "1", "2", "e", "+\infty"),
InitCol1("x'", +1, "x", +1, "y'", +1, "y", +1),
for z in [1, 2, 5, 6] do traitH(z) od, traitV(1),
doubleB(2, 2, NbCol), doubleB(4, 2, NbCol),
valeursLig(2, +1, "-", +2, "-", +1, "0", "+"),
valeursLig(6, +1, "-", +1, "-", "0", +1, "+"),
variationsLig(3, ["\,0", labelstyle :=left],

-2, ["-\infty\,", labelstyle :=right],
2-2*i, jump, ["\,+\infty", labelstyle :=left],
-2+2*i, ["e", linestyle :=dotted, ligdep :=3],
+2, "+\infty" ),

variationsLig(7, ["\,0", labelstyle :=left],
-2, ["-\infty\,", labelstyle :=right],
2-2*i, jump, ["\,+\infty", labelstyle :=left],
-2, ["4", linestyle :=dotted, ligdep :=7],
+2+2*i, "+\infty" )

] ;
\end{texgraph}

−→u −→v

−→w

−→u ∧−→v

θ

Figure 12

\begin{texgraph}[name=figure12, file]
Graph image = [
view(-3, 7, -3, 5), Marges(0, 0, 0, 0), size(8, 1), A :=M(2, 1.5, 0),
theta :=-21*deg, phi :=74*deg,
DrawPlan([A, vecK], vecI, 11, 8, 2),
u :=M(4, 0, 0), v :=3*vecJ, w :=M(2, 1, 3),
Dparallelep(Origin, u, v, w, 0),
Width :=8, Arrows :=1,
Ligne3D([Origin, u, jump, u, u+v, jump, Origin, w, jump, Origin, 4*vecK], 0),
Width :=4, Arrows :=0, Arc3D(vecK, Origin, w, 1.5, 1),
LineStyle :=dashed, Ligne3D([pz(w), w, pxy(w), Origin], 0),
LineStyle :=solid,
angleD(Proj3D(vecK), 0, Proj3D(-vecJ), 0.25),
angleD(0, Proj3D(pxy(w)), Proj3D(w), 0.25),
LabelDot3D(u/2, "$\vec{u}$", "SO"), LabelDot3D(u+v/2, "$\vec{v}$", "S"),
LabelDot3D(w, "$\vec{w}$", "N"), LabelDot3D(4*vecK, "$\vec{u}\wedge\vec{v}$", "N"),
LabelDot3D(1.5*normalize(w+Norm(w)*vecK), "$\theta$", "N")
] ;
\end{texgraph}

Figure 13

\begin{texgraph}[name=figure13, file]
Graph image = [
view3D(-8, 9, -7, 7, -9, 6), view(-8, 8, -8.25, 6.75), Marges(0, 0, 0, 0), size(8, 1),
theta :=20*deg, phi :=74*deg, {background(full, Rgb(0.3, 0.3, 0.3)), }
R :=4, d :=5, r :=3, K :=(R^2-r^2+d^2)/2/d,
S :=Sphere([0, 0], R, 25, 25),
S' :=Sphere([0, -d], r, 25, 25),
P :=[M(0, 0, -K), vecK], P' :=[M(0, 0, -K), -vecK],
L :=Intersection(P, S),
S :=ClipFacet(S, P), S' :=ClipFacet(S', P'),
Build3D(

bdPlan(P, [color :=brown, opacity :=0.3, scale :=0.5]),
bdFacet(S, [color :=steelblue, backculling :=1, smooth :=1]),
bdFacet(S', [color :=forestgreen, backculling :=1, smooth :=1]),
bdAxes([0, 0], [labels :=0])
),

Display3D(),
M :=[sqrt(R^2-K^2)*exp(i*pi/3), -K], H :=[0, -K], O' :=M(0, 0, -d),
Width :=8, LineStyle :=dotted,
Ligne3D([Origin, M, O', jump, M, H], 0),
Arrows :=1, Axes3D(0, 0, 0), Arrows :=0,
LineStyle :=solid, Width :=4, angleD(0, Proj3D(H), Proj3D(M), 0.25),
Color :=red, Width :=8, DrawAretes(L, 0),
Color :=black, DotStyle :=bigdot,
LabelDot(0, "$O$", "NO", 1), LabelDot(Proj3D(M), "$M$", "E", 1),
LabelDot(Proj3D(H), "$H$", "O", 1), LabelDot(Proj3D([0, -d]), "$O'$", "E", 1),
] ;
\end{texgraph}

x −∞ 0 +∞

arctan

−π
2

π
2

0

Figure 14

\begin{texgraph}[name=figure14, file]
Include "variations.mod" ;
Graph image = [
InitLig1("x", "-\infty", "0", "+\infty"),
InitCol1(+1, "\arctan\,\,\,", +1),
variationsLig(4, "-\frac{\pi}2",

+1,
["0", labelstyle :=framed, arrows :=0],
+1,
"\frac{\pi}2"

),
Width :=8, traitH(1), traitV(1)
] ;
\end{texgraph}

x −∞ 0 +∞

argsh

+∞
0

Figure 15

\begin{texgraph}[name=figure15, file]
Include "variations.mod" ;
Graph image = [
InitLig1("x", "-\infty", "0", "+\infty"),
InitCol1(+1, "\argsh\,\,", +1),
variationsLig(4, "-\infty",

+1,
["0", labelstyle :=framed, linestyle :=dotted, arrows :=0],
+1,
"+\infty"

),
Width :=8, traitH(1), traitV(1)
] ;
\end{texgraph}

Figure 16

\begin{texgraph}[name=figure16, file]
Graph image = [
view(-11, 11, -11.5, 10.5), Marges(0, 0, 0, 0), size(8, 1), view3D(-8, 9, -9, 9, -9, 9),
theta :=-40*deg, phi :=68*deg,
A :=M(1, 3, 2), plan :=[A, vecI], u :=-1.5*vecJ, u' :=1.5*M(-1, -1/2, 0),

A' :=M(3, -1, -3)-u', w :=Prodvec(u, u'),
D :=getdroite([A, u]), D' :=getdroite([A', u']),
H' :=interDP([A', u'], plan), H :=interDP([A, u], [A', Prodvec(u', w)]),
D'' := getdroite([H', w]), P :=[px(A), vecI],

Build3D(
bdPlan(P, [color :=beige]),
bdLine([D, jump, D'], [color :=blue]),
bdLine(D'', [color :=red])
),

Display3D(),
Width :=8, LineStyle :=dotted, Color :=blue, Ligne3D(D', 0),
LineStyle :=solid, Color :=black,
angleD( Proj3D(A'), Proj3D(H'), Proj3D(H), 0.25),
angleD( Proj3D(A), Proj3D(H), Proj3D(H'), 0.25),
Color :=black, DotStyle :=bigdot, LineStyle :=solid,
LabelDot( Proj3D(A), "$A$", "N", 1), LabelDot( Proj3D(A'), "$A'$", "N", 1),
LabelDot( Proj3D(H), "$H$", "NO", 1), LabelDot( Proj3D(H'), "$H'$", "SO", 1),
LabelDot( Proj3D(A'+0.5*u'), "$\vec{u'}$", "S", 0),
LabelDot( Proj3D(A+0.5*u), "$\vec{u}$", "S", 0),
LabelDot( Proj3D(H+w), "$\vec{u}\wedge\vec{u'}$", "O", 0),
LabelDot( Proj3D( D[3, 2] ), "$\mathcal{D}$", "E", 0),
LabelDot( Proj3D( D'[1, 2] ), "$\mathcal{D'}$", "S", 0),
LabelDot( Proj3D( D''[3, 2] ), "$\mathcal{D''}$", "N", 0),
Arrows :=1, Ligne3D( [A', A'+u', jump, A, A+u, jump, H, H+w], 0), Arrows :=0
] ;
\end{texgraph}

Figure 17

\begin{texgraph}[name=figure17, file]
Include "variations.mod" ;
Graph image = [
InitLig1("x", "0", "1", "+\infty"),
InitCol1("\ln'", +1, "\ln", +1),
valeursLig(2, +1, "+", +1, "+"),
doubleB(2, 2, NbLig),
variationsLig(5, ["\,-\infty", labelstyle :=left],

+1,
["0", arrows :=0, labelstyle :=framed, linestyle :=solid, ligdep :=2], +1,
["+\infty"]

),
Width :=8, traitH(1), traitH(2), traitV(1)
] ;
\end{texgraph}

O AH

B

Figure 18

\begin{texgraph}[name=figure18, file]
Graph image = [
view(-5, 7, -5, 4), Marges(0, 0, 0, 0), size(8),
O :=-4-4*i, A :=4-4*i, B :=-2+3*i, H :=proj(B, [O, A]), B' :=B+A-O,
Arrows :=1,
Width :=12, Ligne([O, B, jump, O, A], 0), Arrows :=0, Width :=4,
Ligne([B, H], 0), angleD(A, H, B, 0.25),
LineStyle :=dashed, Ligne([A, B, B'], 1),
LabelDot(O, "$O$", "S"), LabelDot(A, "$A$", "S"),
LabelDot(H, "$H$", "S"), LabelDot(B, "$B$", "N")
] ;
\end{texgraph}

Figure 19

\begin{texgraph}[name=figure19, file]
Graph image = [
view(-4, 5, -4, 5), Marges(0, 0, 0, 0), size(8, 1),
Arrows :=1, Axes(0, 1+i), Arrows :=0,
Droite(1, -1, 0), Width :=12, tMin :=0, Cartesienne(ln(x)),
Color :=red, tMin :=Xmin, Cartesienne(exp(x)), Width :=4, Color :=black,
x :=3.25, A :=x+i*ln(x), B :=ln(x)+i*x, C :=(A+B)/2,
DotScale :=1.5, DotStyle :=cross,
Point(A, B), Ligne([A, B], 0),
angleD(0, C, B, 0.15),
markseg(A, C, 2, 0.15, 0.25), markseg(B, C, 2, 0.15, 0.25),
LabelDot(4+i*ln(4), "$\mathcal{C}_{\ln}$", "SE"),
LabelDot(1.5+i*exp(1.5), "$\mathcal{C}_{\exp}$", "SE")
] ;
\end{texgraph}

0

Figure 20

\begin{texgraph}[name=figure20, file]
Include "variations.mod" ;
Graph image = [
InitLig1("\theta", "0", "\frac{\pi}4"),
InitCol1(+1, "\rho", +1),
traitH(1), traitV(1),
variationsLig(2, "a", -2, "0")
] ;
\end{texgraph}

α

Figure 21

\begin{texgraph}[name=figure21, file]
Graph image = [
view(-5, 5, -5, 5), Marges(0, 0, 0, 0), size(6),
Width :=8, Cercle(0, 4), B :=4*exp(-i*pi/5), A :=-bar(B), C :=4*exp(i*115*deg),
Ligne([A, C, B], 0), Droite(A, B),
LabelDot(C, "$M$", "N", 1), LabelDot(A, "$A$", "SO", 1),
LabelDot(B, "$B$", "SE", 1),
Arrows :=1, LabelArc(A, C, B, 1, 1, "$\alpha$")
] ;
\end{texgraph}

Figure 22

\begin{texgraph}[name=figure22, file]
Include "variations.mod" ;
Graph image = [
InitLig1("t", "0", "+\infty"),
InitCol1(+1, "X", +2, "Y", +1),
traitH(1), traitH(4), traitV(1),
variationsLig(4, "a", 2, "+\infty"),
variationsLig(7, "0", 2, "+\infty")
] ;
\end{texgraph}

O A(a) B(b)−→u R

|b −a|

Figure 23

\begin{texgraph}[name=figure23, file]
Graph image = [
view(-5, 5, -1, 2), Marges(0, 0, 0, 0), size(8),
DotStyle :=cross, DotScale :=1.5, A :=0, B :=3.5, O :=-3, u :=2,
Width :=8, Droite(0, 1, 0), Arrows :=1, Width :=12, Ligne([O, O+u], 0), Arrows :=0, Width :=8,
LabelAxe(x, O, "$O$", 1, 1), LabelAxe(x, A, "$A(a)$", 1, 1), LabelAxe(x, B, "$B(b)$", 1, 1),
LabelAxe(x, O+u/2, "$\vec{u}$"), LabelAxe(x, Xmax, "$\rr$"),
LineStyle :=dashed, Ligne([A, A+i, jump, B, B+i], 0), LineStyle :=solid,
Arrows :=2, LabelSeg(A+i, B+i, "$|b-a|$")
] ;
\end{texgraph}

Figure 24

\begin{texgraph}[name=figure24, file]
Graph image = [
view(-1.5, 1.5, -1.5, 1.5), Marges(0, 0, 0, 0), size(6),
Arrows :=1, Axes(0, 1+i), Arrows :=0,
Width :=8, Color :=red, tMin :=-pi, tMax :=pi,
Courbe(exp(i*t)+i*sin(t)*cos(t))
] ;
\end{texgraph}

Figure 25

\begin{texgraph}[name=figure25, file]
Graph image = [
view(-5, 5, -5, 5), Marges(0, 0, 0, 0), size(5),
Width :=8, Cercle(0, 4), B :=4*exp(-i*pi/10), A :=-4*exp(i*pi/5),
C :=4*exp(i*115*deg), Ligne([A, C, B, 0], 1),
LabelDot(C, "$M$", "N", 1), LabelDot(A, "$A$", "SO", 1),
LabelDot(B, "$B$", "SE", 1), LabelDot(0, "$O$", "N", 1),
Arrows :=1, LabelArc(A, C, B, 0.5, 1, "$\alpha$"),
LabelArc(A, 0, B, 0.5, 1, "$2\alpha$")
] ;
\end{texgraph}

Figure 26

\begin{texgraph}[name=figure26, file]
Graph image = [
view(-6.5, 6.5, -5.25, 5.75), Marges(0, 0, 0, 0), size(8, 1),
theta :=33*deg, phi :=74*deg, view3D(-5, 5, -5, 5, -5, 5),
S :=Sphere(Origin, 4, 25, 25), d :=2, x :=sqrt(16-d^2), H :=M(0, d, 0),
D :=getdroite([H, vecI], 1.25),
Build3D(

bdFacet(S, [color :=steelblue, backculling :=1, smooth :=1]),
bdLine(D, [color :=red, width :=12]),
bdAxes([0, 0], [labels :=0, width :=12])
),

Display3D(),
Color :=red, Width :=12, LineStyle :=dotted, Ligne3D(D, 0),
Color :=black, Arrows :=1, Axes3D(0, 0, 0), Arrows :=0, Color :=black, LineStyle :=solid,

Width :=8,
DotStyle :=bigdot,
LabelDot(Proj3D(M(x, d, 0)), "$A$", "SE", 1),
LabelDot(0, "$O$", "NO", 1),
LabelDot(Proj3D(H), "$H$", "S", 1),
LabelDot(Proj3D(M(-x, d, 0)), "$B$", "S", 1),
LabelDot(Proj3D(D[1, 2]), "$\mathcal{D}$", "N"),
Dsphere(Origin, 4, 0)

] ;
\end{texgraph}

0 1−1

0

Figure 27

\begin{texgraph}[name=figure27, file]
Graph image = [
view(-1.5, 1.5, -0.75, 0.75), Marges(0, 0, 0, 0), size(8, 1), LabelSize :=footnotesize,
NewMac("Sqrt", "sqrt(abs(z))*exp(i*Arg(z)/2)", z),
Arrows :=1, Axes(0, 1+i), Arrows :=0,
Width :=8, Color :=red, tMin :=-pi, tMax :=pi,
Polaire(Sqrt(cos(2*t)))
] ;
\end{texgraph}

Figure 28

\begin{texgraph}[name=figure28, file]
Graph image = [
view(-3, 3, -2.5, 2.5), Marges(0, 0, 0, 0), size(6),
Width :=6, Cercle(0, 2), Droite(2, 2+i),
Arrows :=2, Ligne([0, 2], 0), Arrows :=0, DotStyle :=cross,
LabelDot(0, "$O$", "N", 1), LabelDot(2, "$T$", "NE", 1),
angleD(0, 2, 2+i, 0.2),
LabelDot(1, "$d$", "S", 0),
LabelDot( 2*i, "$\mathcal{C}$", "NO", 0),
LabelDot( 2+2*i, "$\mathcal{D}$", "NE", 0),
Label(-2.5*i, "$d=R$")
] ;
\end{texgraph}

Figure 29

\begin{texgraph}[name=figure29, file]
Graph image = [
view(0, 5, 0, 5), Marges(0.25, 0, 0, 0.5), size(5, 0),
Arrows :=1, Axes(0, 1+i, 1+i), Arrows :=0, Droite(1, -1, 0),
Width=8, Cartesienne(ch(x)),
LabelDot(2.25+ch(2.25)*i, "$\mathcal{C}_{\ch}$", "SE"),
Width :=8, Color :=red, Cartesienne(argch(x)),
LabelDot(4+argch(4)*i, "$\mathcal{C}_{\argch}$", "SE"),
Color :=black, Arrows :=1,
Ligne([i, i+0.5, jump, 1, 1+0.5*i], 0)
] ;
\end{texgraph}

Figure 30

\begin{texgraph}[name=figure30, file]
Graph image = [
view(-1.5, 1.5, -1.5, 1.5), Marges(0, 0, 0, 0), size(6, 1),
Axes(0, 0), Cercle(0, 1), Droite(1, 0, 1), x :=40*deg, A :=exp(i*x),
Width :=12, Arrows :=1, Ligne([0, i, jump, 0, 1, jump, 1, 1+i], 0), Arrows :=0, Width :=4,
LineStyle :=dashed, Ligne([i*sin(x), A, cos(x)], 0), LineStyle :=solid,
Ligne([0, 1+i*tan(x)], 0),
Arrows :=1, LabelArc(1, 0, A, 1, 1, "$x$"), Arrows :=0,
DotStyle :=cross, DotScale :=1.5,
LabelDot(i*sin(x), "$\sin(x)$", "O", 1), LabelDot(A, "$M$", "N", 1),
LabelDot(cos(x), "$\cos(x)$", "S", 1), LabelDot(0, "$O$", "SO"),
LabelDot(1+i*tan(x), "$\tan(x)$", "E", 1)
] ;
\end{texgraph}

Figure 31

\begin{texgraph}[name=figure31, file]
Graph image = [
view(-1, 4, -1, 3), Marges(0, 0, 0, 0), size(6, 0), DotStyle :=cross,
Axes(0, 0), A :=3.5+2.5*i,
LineStyle :=dashed, Ligne([i*Im(A), A, Re(A)], 0), LineStyle :=solid,
Arrows :=1, Width :=8,
Ligne([0, 1, jump, 0, i, jump, 0, A], 0),
Arrows :=0,
LabelAxe(x, 0.5, "$\vec{\imath}$"),
LabelAxe(y, i*0.5, "$\vec{\jmath}$"),
LabelAxe(x, Re(A), "$x(t)$", 1, 1),
LabelAxe(y, i*Im(A), "$y(t)$", 2, 1),
LabelDot(0, "$O$", "SO"),
LabelDot(A/2, "$\vec{f}(t)$", "E", 0, 0.35)
] ;
\end{texgraph}

Figure 32

\begin{texgraph}[name=figure32, file]
Graph image = [
view(-3, 3, -3.75, 3.75), Marges(0, 0, 0, 0), size(8, 1),
LabelSize :=footnotesize,
Axes(0, 0), A :=9/4-1+3*i, d :=-2, H :=proj(A, [d, d+i]), p :=2,
Arrows :=1, Width :=8, Ligne([0, 1, jump, 0, i], 0), Width :=4,
LabelArc(1, 0, A, 1.25, 1, "$t$"), Arrows :=0,
Ligne([H, A, 0], 0), Droite(d, d+i), DotScale :=1.15,
LabelDot(A, "$M$", "N", 1),
LabelAxe(x, 0.5, "$\vec{\imath}$", 1), LabelAxe(y, 0.5*i, "$\vec{\jmath}$", 2),
LabelDot(d, "$-d$", "SO", 1), LabelDot(0, "$F$", "SO", 1), LabelDot(d/2, "$O$", "SO", 1),
LabelDot(H, "$H$", "O", 1),
LabelDot(d+Ymin*i, "$\mathcal{D}$", "NO"),
Seg(H, 0), Dmed(0, H, 1), markseg(H, H/2, 2, 0.125, 0.25), markseg(0, H/2, 2, 0.125, 0.25),
Droite(A, A-H), angleD(H/2, A, A+H, 0.25),
Width :=8, Color :=red, tMin :=-4, tMax :=4, Courbe(t^2/4-1+i*t),
Color :=darkblue, Ligne([Xmax+i*Im(A), A, 0], 0),
flecher( [Xmax+i*Im(A), A], 0.5), flecher( [A, 0], 0.5)
] ;
\end{texgraph}

Figure 33

\begin{texgraph}[name=figure33, file]
Graph image = [
view(-3, 3, -2.5, 2.5), Marges(0, 0, 0, 0), size(6),
Width :=6, Cercle(0, 2), Droite(1.5, 1.5+i),
Arrows :=2, Ligne([0, 1.5], 0), Arrows :=0, DotStyle :=cross,
LabelDot(0, "$O$", "N", 1), y :=sqrt(4-1.5^2),
LabelDot(1.5+i*y, "A", "E", 1), LabelDot(1.5-i*y, "B", "E", 1),
LabelDot(0.75, "$d$", "S", 0),
LabelDot( 2*i, "$\mathcal{C}$", "NO", 0),
LabelDot( 1.5+2*i, "$\mathcal{D}$", "NE", 0),
Label(-2.5*i, "$d<R$")
] ;
\end{texgraph}

Figure 34

\begin{texgraph}[name=figure34, file]
Include "draw2d.mod" ;
Graph image = [
view(-5, 5, -4.5, 3), Marges(0, 0, 0, 0), size(8),
Arrows :=1, Axes(0, 1+i), Arrows :=0,
Width :=2, LineStyle :=dashed, Color :=black,

Ligne([-4-3*i, -4+i*sin(-4), jump, -1-i*sin(1), -1+2*i, jump,
1-i, 1+i, jump, 4, 4-4*i], 0),

Width :=8, LineStyle :=solid, Color :=red,
tMin :=-4, tMax :=1, NbPoints :=50,
draw("cartesian", if x <=-1 then sin(x) else x^2 fi,

[discont :=1, marker :=[0, Oarc, 1, Oarc]]),
tMin :=1, tMax :=4,
draw("cartesian", -(x-2)^2, marker :=[0, Carc, 1, Carc]),
DotStyle :=dot, Point(-4-3*i, -1+2*i)
] ;
\end{texgraph}

Figure 35

\begin{texgraph}[name=figure35, file]
Graph image = [
Marges(0, 0, 0, 0), view(-1.5, 2.5, -1.5, 2.5), size(5),
Width :=4, Arrows :=1, Axes(0, 1+i), Arrows :=0,
Width :=6, Color :=blue, Cercle(0, 1), A :=2.25+2*i, B :=A/abs(A),
Color :=red, Ligne([0, A], 0), Color :=black, DotScale :=1.5,
LabelArc(1, 0, A, 0.75, 1, "$\theta$", [Arrows :=1]),
LabelDot(A, "$A(z)$", "N", 1), LabelDot(B, "$B(\frac{z}{|z|})$", "E", 1)
] ;
\end{texgraph}

Figure 36

\begin{texgraph}[name=figure36, file]
Graph image = [
view(-0.5, 5, -3.5, 2), Marges(0, 0, 0, 0), size(5, 1),
Arrows :=1, Axes(0, 1+i), Arrows :=0,
Width :=8, Color :=red,
Cartesienne(ln(x)),
Label(1+i/2, "$\mathcal{C}_{\ln}$")
] ;
\end{texgraph}

Figure 37

\begin{texgraph}[name=figure37, file]
Graph image = [
view(-6, 6, -4, 4), Marges(0, 0, 0, 0), size(8, 1), LabelSize :=footnotesize,
a :=3, b :=2, c :=sqrt(a^2+b^2),
F :=c, F' :=-c, t1 :=1.15, A :=a*ch(t1)+b*i*sh(t1), d :=a^2/c,
H' :=proj(A, [-d, -d+i]), H :=proj(A, [d, d+i]),
Axes(0, 0), Droite(-d, -d+i), Droite(d, d+i), Ligne([H', A, F', jump, A, F], 0),
Color :=blue, Droite(b, -a, 0), Droite(b, a, 0),
LabelDot((Ymin+0.25)*(-a/b+i), "$Y=-\frac abX$", "O", 0, 0.5),
LabelDot((Ymin+0.25)*(a/b+i), "$Y=\frac abX$", "E", 0, 0.5),
Color :=black,
angleD(H', H, H+i, 0.25), angleD(H, H', H'+i, 0.25), DotScale :=1.15,
Width :=8, Arrows :=1, Ligne([0, 1, jump, 0, i], 0), Arrows :=0,
Color :=red, tMin :=-3, tMax :=3, Courbe(a*ch(t)+i*b*sh(t)), Courbe(-a*ch(t)+i*b*sh(t)),
Width :=4, Color :=black,
LabelDot(H, "$H$", "NE", 1), LabelDot(H', "$H'$", "O", 1),
LabelDot(d+i*Ymax, "$\mathcal{D}$", "SO"),
LabelDot(-d+i*Ymax, "$\mathcal{D}'$", "SE"), LabelDot(A, "$M$", "SE", 1),
LabelDot(F, "$F(c,0)$", "SE", 1), LabelDot(F', "$F'(-c,0)$", "SO", 1),
LabelDot(-d, "$-\frac{a^2}c$", "SE", 1), LabelDot(d, "$\frac{a^2}c$", "SO", 1),
DotStyle :=cross, LabelDot(a, "$a$", "NE", 1, 0.1),
LabelStyle :=framed, LineStyle :=noline, FillStyle :=full, DotStyle :=dot,
LabelDot(0, "$O$", "S", 1)
] ;
\end{texgraph}

\begin{texgraph}[name=figure38, file]
Include "variations.mod" ;
Graph image = [
InitLig1("t", "0", "\frac{\pi}2"), InitCol1(+1, "X", +2, "Y", +1),
traitH(1), traitH(4), traitV(1),
variationsLig(2, "a", -2, "0"), variationsLig(7, "0", 2, "b")
] ;
\end{texgraph}

Figure 39

\begin{texgraph}[name=figure39, file]
Include "variations.mod" ;
Graph image = [
dimTab(4, 4), defLargeCol(0, 1.5), defHautLig(0, 1.5),
InitLig1("\theta", "-\frac{\pi}2", "\frac{\pi}2"),
InitCol1(+1, "\rho", +1),
traitH(1), traitV(1), doubleB(4, 2),
variationsLig(2, "\ln(2)", -2, ["-\infty\,", labelstyle :=right])
] ;
\end{texgraph}

Figure 40

\begin{texgraph}[name=figure40, file]
Graph image = [
view(-7, 7, -5, 5, ), Marges(0, 0, 0, 0), size(10),
SaveWin(), SaveAttr(),
view(-7, 1, 0.5, 5), ChangeWinTo([-1.25*i-4, 4+1.25*i]),
Arrows :=1, axeY(0, 1), axeX(0, pi/2, 1, bottom, 1, "\pi", 2, 0), Arrows :=0,
LineStyle :=dashed, Ligne([i, pi/2+i, pi/2], 0), LineStyle :=solid,
Color :=red, Width :=8, Cartesienne(sin(x)),
LabelDot(-3.5-sin(3.5)*i, "$\mathcal{C}_{\sin}$", "E"),
Color :=black, Width :=4, Arrows :=2,
tangente(sin(x), [-pi/2, pi/2], 1), x :=Nil,
RestoreAttr(), RestoreWin(),

SaveWin(), SaveAttr(),
view(-7, 1, -5, -0.5), ChangeWinTo([-1.25*i-4, 4+1.25*i]),
Arrows :=1, axeY(0, 1), axeX(0, pi/2, 1, bottom, 1, "\pi", 2, 0), Arrows :=0,
LineStyle :=dashed, Ligne([-i, pi-i, pi], 0), LineStyle :=solid,
Color :=red, Width :=8, Cartesienne(cos(x)),
LabelDot(0.75+cos(0.75)*i, "$\mathcal{C}_{\cos}$", "E"),
Color :=black, Width :=4, Arrows :=2,
tangente(cos(t), [-pi, 0, pi], 1),
RestoreAttr(), RestoreWin(),
SaveWin(), SaveAttr(),
view(2, 7, -5, 5), ChangeWinTo([-5*i-2, 2+5*i]),
Arrows :=1, axes(0, i), Arrows :=0, tMin :=-pi/2, tMax :=pi/2,
LineStyle :=dashed, Droite(1, 0, -pi/2), Droite(1, 0, pi/2), LineStyle :=solid,
LabelDot(pi/2, "$\frac{\pi}2$", "SO"), LabelDot(-pi/2, "$-\frac{\pi}2$", "SE"),
Color :=red, Width :=8, Cartesienne(tan(x)),
LabelDot(0.5+tan(0.5)*i, "$\mathcal{C}_{\tan}$", "E"),
RestoreAttr(), RestoreWin()
] ;
\end{texgraph}

Figure 41

\begin{texgraph}[name=figure41, file]
Include "variations.mod" ;
Graph image = [
dimTab(9, 6), defHautLig(7, 2),
InitLig1("t", "0", "\frac{\pi}3", "\pi"),
InitCol1("x'", +1, "x", +1, "y'", +1, "y", +1),
traitV(1), traitH(1), traitH(2), traitH(5),
traitH(6),
valeursLig(2, "0", +1, "-", +1, "0"),
valeursLig(6, "2", "+", "0", "-", "0"),
variationsLig(3, "1", -2+2*i, "-1"),
variationsLig(9, "0", +2, "\frac{3\sqrt{3}}4", -2, "0")
] ;
\end{texgraph}

Figure 42

\begin{texgraph}[name=figure42, file]
Graph image = [
view(-3, 3, -4, 4), Marges(0, 0, 0, 0), size(8, 1),
LabelSize :=footnotesize,
Axes(0, 0), A :=9/4-1+3*i, d :=-2, H :=proj(A, [d, d+i]), p :=2,
Arrows :=1, Width :=8, Ligne([0, 1, jump, 0, i], 0), Width :=4,
LabelArc(1, 0, A, 1.25, 1, "$t$"), Arrows :=0,
Ligne([H, A, 0], 0), Droite(d, d+i), DotScale :=1.15,
LabelDot(A, "$M$", "E", 1),
LabelAxe(x, 0.5, "$\vec{\imath}$", 1), LabelAxe(y, 0.5*i, "$\vec{\jmath}$", 2),
LabelDot(d, "$-d$", "SO", 1), LabelDot(0, "$F$", "SO", 1), LabelDot(d/2, "$O$", "SO", 1),
LabelDot(H, "$H$", "O", 1),
LabelDot(d+Ymin*i, "$\mathcal{D}$", "NO"),
Width :=8, Color :=red, tMin :=-4, tMax :=4, Courbe(t^2/4-1+i*t)
] ;
\end{texgraph}

\begin{texgraph}[name=figure43, file]
Graph image = [
view(-4, 7, -4, 7), Marges(1, 0, 0, 0), size(6, 0), LabelSize :=footnotesize,
Arrows :=1, Axes(0, 1+i), Arrows :=0,
Droite(1, -1, -0.5), Width :=8, Color :=red,
tMin :=0, tMax :=20,
Courbe(t/ln(t)+i*t*t/(t-1))
] ;
\end{texgraph}

Figure 44

\begin{texgraph}[name=figure44, file]
Graph image = [
view(-3, 3, 0, 5), Marges(0, 0, 0, 0.5), size(6),
Arrows :=1, Axes(0, 1+i, 1+2*i), Arrows :=0,
Width :=8, Color :=red, Cartesienne(ch(x)),
LabelDot(2+ch(2)*i, "$\mathcal{C}_{\ch}$", "SE"),
Color :=black, Arrows :=2, tangente(ch(x), 0, 2)
] ;
\end{texgraph}

\begin{texgraph}[name=figure45, file]
Graph image = [
view(-3.5, 7.5, -3, 6), Marges(0, 0, 0, 0), size(8, 1),
theta :=-21*deg, phi :=74*deg, A :=M(0, 0, 5), B :=M(3, 3, 0),
DrawPlan([B/2, vecK], vecI, 8, 8, -4),
Ligne3D([A, Origin, B], 1),
angleD(Proj3D(A), 0, Proj3D(B), 0.25),
LabelDot3D(A, "$M$", "N", 1), LabelDot3D(B, "$A$", "E", 1), LabelDot3D(pxy(A), "$H$", "O",
1),
LabelDot3D(M(6, 0, 0), "$\mathcal{P}$", "S")
] ;
\end{texgraph}

Figure 46

\begin{texgraph}[name=figure46, file]
Graph image = [
view(-4, 6, -4, 5), Marges(0, 0, 0, 0), size(8),
x :=-2,
setB("B", x, [scale :=0.5]),
setB("A", x, [scale :=0.25, rotation :=15]),
Label(-2-2.75*i, "$A\subset B$"),
x :=2,
setB("B", x+1.5, [scale :=0.5, rotation :=-15]),
setB("A", x, [scale :=0.5, rotation :=15]),
Label(x+0.75-2.75*i, "$A\not\subset B$")
] ;
\end{texgraph}

\begin{texgraph}[name=figure47, file]
Graph image = [
Marges(0, 0, 0.25, 0.25), size(7),
Width :=8, Arrows :=1, Axes(0, 0), Arrows :=0,
Cercle(0, 4), alpha :=40*deg, u :=4*exp(i*alpha), K :=-0.5,
Droite(0, u), Droite(0, i*u),
Width :=12, Arrows :=1, Ligne([0, u, jump, 0, i*u], 0),
Width :=8, LabelArc(1, 0, u, 1.5, 1, "$\theta_0$"), Arrows :=0,
angleD(u, 0, i*u, 0.5),
LabelDot(u, "$\vec{u}(\theta_0)$", "O"),
LabelDot(i*u/2, "$\vec{v}(\theta_0)$", "O"),
LabelDot(K*i*u, "$K$", "N"),
Color :=red,
LabelDot(K*i*u, "$\mathcal{D}$", "S", 0, 0.75), Droite(K*i*u, K*i*u+u)
] ;
\end{texgraph}

D

Figure 48

\begin{texgraph}[name=figure48, file]
Graph image = [
Marges(0, 0, 0, 0),
O :=0, O' :=2, R :=4, R' :=1.5, a :=abs(O'-O), d :=(R^2-R'^2+a^2)/(2*a),
view(-0.5-R, 1.5+R, -0.5-R, 0.5+R), size(8),
Width :=6, Cercle(O, R), Cercle(O', R'), Droite(d, d+i), Droite(O, O'),
Arrows :=2, Ligne([O+i*Ymin, O'+i*Ymin], 0), Arrows :=0,
LineStyle :=dashed, Ligne( [O, O+i*Ymin, jump, O'+i*Ymin, O'], 0),
LineStyle :=solid, DotStyle :=cross,
LabelDot(O, "$O$", "N", 1), LabelDot(O', "$O'$", "N", 1),
SaveAttr(), FillStyle :=full, LineStyle :=noline, LabelStyle :=framed,
Label( (O+O')/2+i*Ymin, "$a$" ), RestoreAttr(),
LabelDot( O+R*i, "$\mathcal{C}$", "NO", 0),
LabelDot( O'+R'*i, "$\mathcal{C'}$", "NE", 0),
LabelDot( d+R*i, "$\mathcal{D}$", "NE", 0)
] ;
\end{texgraph}

Figure 49

\begin{texgraph}[name=figure49, file]
Graph image = [
view(-5, 5, -4, 4), Marges(0, 0, 0, 0), size(8, 1), LabelSize :=footnotesize,
a :=3.5, b :=2, c :=sqrt(a^2-b^2),
F :=-c, F' :=c, t1 :=65*deg, A :=a*cos(t1)+b*i*sin(t1), d :=a^2/c,
H :=proj(A, [-d, -d+i]), H' :=proj(A, [d, d+i]),
Axes(0, 0), Droite(-d, -d+i), Droite(d, d+i), Ligne([H, H', jump, F, A, F'], 0),
angleD(H', H, H+i, 0.25), angleD(H, H', H'+i, 0.25), DotScale :=1.15,
Width :=8, Arrows :=1, Ligne([0, 1, jump, 0, i], 0), Arrows :=0,
Color :=red, Ellipse(0, a, b), Width :=4, Color :=black,
LabelDot(H, "$H$", "O", 1), LabelDot(H', "$H'$", "E", 1),
LabelDot(-d+i*Ymax, "$\mathcal{D}$", "SE"), LabelDot(d+i*Ymax, "$\mathcal{D}'$", "SO"),
LabelDot(A, "$M$", "N", 1),
LabelDot(F, "$F(-c,0)$", "SE", 1), LabelDot(F', "$F'(c,0)$", "SO", 1),
LabelDot(-d, "$-\frac{a^2}c$", "SO"), LabelDot(d, "$\frac{a^2}c$", "SE", 1),
DotStyle :=cross,
LabelDot(a, "$a$", "SE", 1), LabelDot(b*i, "$b$", "NO", 1)
] ;
\end{texgraph}

Figure 50

\begin{texgraph}[name=figure50, file]
Graph image = [
Marges(0, 0, 0.5, 0.5),
O :=0, O' :=5, R :=4, R' :=2.5, a :=abs(O'-O), d :=(R^2-R'^2+a^2)/(2*a),
view(-0.5-R, O'+0.5+R', -0.5-R, 0.5+R), size(8),
Width :=6, Cercle(O, R), Cercle(O', R'), Droite(d, d+i), Droite(O, O'),
Arrows :=2, Ligne([O+i*Ymin, O'+i*Ymin], 0), Arrows :=0,
LineStyle :=dashed, Ligne( [O, O+i*Ymin, jump, O'+i*Ymin, O'], 0),
LineStyle :=solid, DotStyle :=cross,
y :=sqrt(R^2-d^2),
LabelDot(d+i*y, "$A$", "O", 1), LabelDot(d-i*y, "$B$", "O", 1),
LabelDot(O, "$O$", "N", 1), LabelDot(O', "$O'$", "N", 1),
SaveAttr(), FillStyle :=full, LineStyle :=noline, LabelStyle :=framed,
Label( (O+O')/2+i*Ymin, "$a$" ), RestoreAttr(),
LabelDot( O+R*i, "$\mathcal{C}$", "NO", 0),
LabelDot( O'+R'*i, "$\mathcal{C'}$", "NE", 0),
LabelDot( d+R*i, "$\mathcal{D}$", "NE", 0)
] ;
\end{texgraph}

Figure 51

\begin{texgraph}[name=figure51, file]
Graph image = [
view(-5, 5, -3, 3), size(6), Width :=8, FillColor :=lightgray,
FillStyle :=full, DotStyle :=times, DotScale :=1.5,
x :=-3,
setB("A", x, [scale :=0.5]), LabelDot(x, "$x$", "O", 1),
Label(x-2.5*i, "$x\in A$"),
x :=2,
setB("A", x, [scale :=0.5]), LabelDot(x+2, "$x$", "E", 1),
Label(x-2.5*i, "$x\notin A$")
] ;
\end{texgraph}

A B

A∪B

Figure 52

\begin{texgraph}[name=figure52, file]
Graph image = [
view(-3, 3, -3, 3), Width :=8, size(6),
A :=setB("A", -1, [scale :=0.5, rotation :=15]),
B :=setB("B", 1, [scale :=0.5, rotation :=-15]),
C :=cupB(A, B),
FillColor :=lightgray, FillStyle :=full, FillOpacity :=0.8,
Color :=crimson, Bezier(C), LabelStyle :=framed, Label(-2.5*i, "$A\cup B$")
] ;
\end{texgraph}

Figure 53

\begin{texgraph}[name=figure53, file]
Graph image = [
Marges(0, 0, 0, 0),
O :=0, O' :=2.5, R :=4, R' :=1.5, a :=abs(O'-O), d :=(R^2-R'^2+a^2)/(2*a),
view(-0.5-R, 1.5+R, -0.5-R, 0.5+R), size(8),
Width :=6, Cercle(O, R), Cercle(O', R'), Droite(d, d+i), Droite(O, O'),
Arrows :=2, Ligne([O+i*Ymin, O'+i*Ymin], 0), Arrows :=0,
LineStyle :=dashed, Ligne( [O, O+i*Ymin, jump, O'+i*Ymin, O'], 0),
LineStyle :=solid, DotStyle :=cross,
LabelDot(O, "$O$", "N", 1), LabelDot(O', "$O'$", "N", 1),
SaveAttr(), FillStyle :=full, LineStyle :=noline, LabelStyle :=framed,
Label( (O+O')/2+i*Ymin, "$a$" ), RestoreAttr(),
LabelDot( O+R*i, "$\mathcal{C}$", "NO", 0),
LabelDot( O'+R'*i, "$\mathcal{C'}$", "NE", 0),
LabelDot( d+R*i, "$\mathcal{D}$", "NE", 0)
] ;
\end{texgraph}

Figure 54

\begin{texgraph}[name=figure54, file]
Graph image = [
view(-3, 5, -3, 5), view3D(-1, 5, -1, 5, -1, 5), Marges(0, 0, 0, 0), size(8, 1),
theta :=20*deg, phi :=60*deg,
Ligne3D([ M(-1, 0, 0), M(5, 0, 0), jump, M(0, -1, 0), M(0, 5, 0), jump, M(0, 0, -1), M(0, 0, 5)], 0),
A :=M(3, 3, 4), H :=pxy(A), O :=Origin,
Arrows :=1, Width :=12,
Ligne3D([O, M(1, 0, 0), jump, O, M(0, 1, 0), jump, O, M(0, 0, 1), jump, O, A], 0),
Width :=4, Arc3D(M(1, 0, 0), O, H, 1.5, 1), Arrows :=0, Arc3D(vecK, O, A, 1.5, 1),
Ligne3D([O, H], 0), LineStyle :=dashed,
Ligne3D([px(A), H, A, pz(A), jump, H, py(A)], 0),
LabelSize :=footnotesize,
LabelDot3D(py(A), "$y$", "N", 1), LabelDot3D(px(A), "$x$", "O", 1),
LabelDot3D(pz(A), "$z$", "O", 1), LabelDot3D(A, "$A$", "E"), LabelDot3D(H, "$H$", "S"),
LabelDot3D([1.5*exp(i*angle(vecI, H)/2), 0], "$\varphi$", "S"),
LabelDot3D(1.5*normalize(Norm(A)*vecK+A), "$\theta$", "NE"),
LabelDot3D(vecI, "$\vec{u}$", "O"), LabelDot3D(vecJ, "$\vec{v}$", "N"),
LabelDot3D(vecK, "$\vec{w}$", "O")
] ;
\end{texgraph}

Figure 55

\begin{texgraph}[name=figure55, file]
Include "draw2d.mod" ;
Graph image = [
view(-4.5, 4.5, -4.5, 4.5), Marges(0, 0, 0, 0), size(6),
Width :=4, Arrows :=1, Axes(0, 1+i), Arrows :=0,
draw("cartesian", Ent(x), [marker :=[0, Carc, 1, Oarc], Color :=red, Width :=8, discont :=1] ),
Width :=2, LineStyle :=dashed,
Ligne( for k from -4 to 3 do k, k+k*i, jump, k+1, k+1+k*i, jump od, 0)
] ;
\end{texgraph}

Figure 56

\begin{texgraph}[name=figure56, file]
Graph image = [
Marges(0, 0, 0, 0),
O :=0, O' :=2, R :=4, R' :=1.5, a :=abs(O'-O), d :=(R^2-R'^2+a^2)/(2*a),
view(-0.5-R, 1.5+R, -0.5-R, 0.5+R), size(8),
Width :=6, Cercle(O, R), Cercle(O', R'), Droite(d, d+i), Droite(O, O'),
Arrows :=2, Ligne([O+i*Ymin, O'+i*Ymin], 0), Arrows :=0,
LineStyle :=dashed, Ligne( [O, O+i*Ymin, jump, O'+i*Ymin, O'], 0),
LineStyle :=solid, DotStyle :=cross,
LabelDot(O, "$O$", "N", 1), LabelDot(O', "$O'$", "N", 1),
SaveAttr(), FillStyle :=full, LineStyle :=noline, LabelStyle :=framed,
Label( (O+O')/2+i*Ymin, "$a$" ), RestoreAttr(),
LabelDot( O+R*i, "$\mathcal{C}$", "NO", 0),
LabelDot( O'+R'*i, "$\mathcal{C'}$", "NE", 0),
LabelDot( d+R*i, "$\mathcal{D}$", "NE", 0)
] ;
\end{texgraph}

Figure 57

\begin{texgraph}[name=figure57, file]
Graph image = [
Marges(0, 0, 0, 0), view(-2.5, 5.5, -2.5, 2.5), size(8),
O :=0, O' :=4, R :=2, R' :=1.5, a :=abs(O'-O), d :=(R^2-R'^2+a^2)/(2*a),
Width :=6, Cercle(O, R), Cercle(O', R'), Droite(d, d+i), Droite(O, O'),
Arrows :=2, Ligne([O+i*Ymin, O'+i*Ymin], 0), Arrows :=0,
LineStyle :=dashed, Ligne( [O, O+i*Ymin, jump, O'+i*Ymin, O'], 0),
LineStyle :=solid, DotStyle :=cross,
LabelDot(O, "$O$", "N", 1), LabelDot(O', "$O'$", "N", 1),
SaveAttr(), FillStyle :=full, LineStyle :=noline, LabelStyle :=framed,
Label( (O+O')/2+i*Ymin, "$a$" ), RestoreAttr(),
LabelDot( O+R*i, "$\mathcal{C}$", "NO", 0),
LabelDot( O'+R'*i, "$\mathcal{C'}$", "NE", 0),
LabelDot( d+R*i, "$\mathcal{D}$", "NE", 0)
] ;
\end{texgraph}

Figure 58

\begin{texgraph}[name=figure58, file]
Graph image = [
Marges(0, 0, 0, 0), view(-1, 4, -1, 3.5),
Width :=4, Axes(0, 0), M :=1.5*i+3,
LabelDot(0, "$O$", "SO", 0), DotStyle :=cross, LabelDot(M, "$M(x,y)$", "E", 1),
LineStyle :=dashed, Ligne([ i*Im(M), M, Re(M)], 0), LineStyle :=solid,
Ligne([0, M], 0),
Arrows :=1, Width :=8, Ligne([0, 1, jump, 0, i], 0),
Width :=4, LabelArc(1, 0, M, 2, 1, "$\theta$"),
LabelDot(0.5, "$\vec{u}$", "S", 0), LabelDot(0.5*i, "$\vec{v}$", "O", 0),
LabelDot(Re(M), "$x$", "S", 1), LabelDot(i*Im(M), "$y$", "O", 1),
//LabelDot(2+0.5*i, "$\theta$", "E", 0),
LabelStyle :=stacked+top,
Label(3.25*i+1.75, "$x+iy=re^{i\theta}$ avec\\$r=\sqrt{x^2+y^2}=OM$"),
LabelAngle :=Arg(M)*rad, LabelStyle :=bottom, Label(M/2+0.1*i, "$r=OM$", 0)
] ;
\end{texgraph}

Figure 59

\begin{texgraph}[name=figure59, file]
Graph image = [
view(-1.5, 1.5, -5, 0.5), Marges(0, 0, 0, 0), size(6, 1),
LabelSize :=footnotesize,
Arrows :=1, Axes(0, 1+i), Arrows :=0,
Width :=8, Color :=red, tMin :=-pi, tMax :=pi,
Polaire(ln(1-sin(t)))
] ;
\end{texgraph}

Figure 60

\begin{texgraph}[name=figure60, file]
Graph image = [
Marges(0, 0, 0, 0), view(-1, 3, -1, 2.5), Axes(0, 0), M :=1.5*i+2,
LabelDot(0, "$O$", "SO", 0), DotStyle :=cross, LabelDot(M, "$M(x,y)$", "N", 1),
LineStyle :=dashed, Ligne([ i*Im(M), M, Re(M)], 0), LineStyle :=solid,
LabelDot(0.5, "$\vec{u}$", "S", 0), LabelDot(0.5*i, "$\vec{v}$", "O", 0),
LabelDot(Re(M), "$x$", "S", 1), LabelDot(i*Im(M), "$y$", "O", 1),
Width :=4, Arrows :=1, Ligne([0, 1, jump, 0, i], 0)
] ;
\end{texgraph}

Figure 61

\begin{texgraph}[name=figure61, file]
Graph image = [
view(-2, 2, -2, 2), Marges(0, 0, 0, 0), size(8, 0),
Arrows :=1, Axes(0, 0), Arrows :=0,
Droite(1, -1, 0),
LineStyle :=dashed,
Ligne([i, i+pi/2, pi/2, jump, 1, 1+i*pi/2, i*pi/2, jump,

-i, -i-pi/2, -pi/2, jump, -1, -1-i*pi/2, -i*pi/2], 0),
LineStyle :=solid, tMin :=-pi/2, tMax :=pi/2,
Width :=8, Cartesienne(sin(x)),
LabelDot(pi/2+i, "$\mathcal{C}_{\sin}$", "E"),
DotStyle :=cross, DotScale :=1.5,
LabelAxe(x, 1, "$1$", 1, 1), LabelAxe(x, -1, "$-1$", 2, 1),
LabelAxe(x, pi/2, "$\frac{\pi}2$", 1, 1), LabelAxe(x, -pi/2, "$-\frac{\pi}2$", 2, 1),
LabelAxe(y, i, "$1$", 2, 1), LabelAxe(y, -i, "$-1$", 1, 1),
LabelAxe(y, i*pi/2, "$\frac{\pi}2$", 2, 1), LabelAxe(y, -i*pi/2, "$-\frac{\pi}2$", 1, 1),
tMin :=-1, tMax :=0.9999,
Width :=8, Color :=red, Cartesienne(arcsin(x)),
LabelDot(1+pi/2*i, "$\mathcal{C}_{\arcsin}$", "N")
] ;
\end{texgraph}

Figure 62

\begin{texgraph}[name=figure62, file]
Graph image = [
view(-4, 4, -4, 4), Marges(0, 0, 0, 0), size(6, 0),
Arrows :=1, Axes(0, 1+i), Arrows :=0, Droite(1, -1, 0),
Width :=8, Cartesienne(sh(x)),
LabelDot(2+sh(2)*i, "$\mathcal{C}_{\sh}$", "NO"),
Width :=8, Color :=red, Cartesienne(argsh(x)),
LabelDot(3+argsh(3)*i, "$\mathcal{C}_{\argsh}$", "SE"),
] ;
\end{texgraph}

Figure 63

\begin{texgraph}[name=figure63, file]
Graph image = [
view(-3, 3, -3, 3), Marges(0, 0, 0, 0), size(6, 0),
Arrows :=1, Axes(0, 1+i), Arrows :=0, Droite(1, -1, 0),
LineStyle :=dashed, Droite(0, 1, 1), Droite(0, 1, -1), Droite(1, 0, 1), Droite(1, 0, -1),
LineStyle :=solid,
Width :=8, Cartesienne(th(x)),
LabelDot(2+th(2)*i, "$\mathcal{C}_{\th}$", "SE"),
Width :=8, Color :=red, Cartesienne(argth(x)),
LabelDot(1+2.5*i, "$\mathcal{C}_{\argth}$", "E"),
] ;
\end{texgraph}

Figure 64

\begin{texgraph}[name=figure64, file]
Graph image = [
view(-3, 3, -3, 3), Width :=8, size(6),
FillColor :=lightgray, FillStyle :=full,
E :=setB("E", 0, [scale :=0.6, rotation :=40]), FillColor :=white,
A :=setB("A", 0, [scale :=0.25, rotation :=15]),
FillStyle :=none, Color :=crimson, Bezier(E), Bezier(A),
FillStyle :=full, FillColor :=lightgray, LabelStyle :=framed,
Label(-2.75*i, "$\mathrm{C}_E(A)$")
] ;
\end{texgraph}

Figure 65

\begin{texgraph}[name=figure65, file]
Cmd [view(-4.5, 4.5, -3, 4.5), Marges(0, 0, 0, 0), size(8)] ;
Mac

f = if %1<-4 then -2
elif %1<-2 then 3
elif %1<1 then 1
elif %1<3 then -2
else 4
fi ;

Graph objet1 =
[Width :=8, LineStyle :=dotted, FillStyle := bdiag, FillColor :=black,
domaine1(\f(t), -4, 4, 5, 1),
Width :=18, LineStyle :=solid, FillStyle :=none, tMin :=-4, tMax :=4, Color :=red,
Cartesienne(\f(x), 5, 1),
Width :=8, LineStyle :=dotted, Color :=black, Ligne([-2, -2+i], 0),
Width :=4, Arrows :=1, LineStyle :=solid, Axes(0, 1+i, 2+2*i)
] ;

\end{texgraph}

Figure 66

\begin{texgraph}[name=figure66, file]
Graph image = [
view(-5, 5, -2, 5), Marges(0, 0, 0, 0), size(8),
SetMatrix([0, 1, 1+i]), Width :=8,
axes(0, 0, 0, 0), A :=-4.5+3*i,
Width :=12, Arrows :=1, Ligne( [0, 3, jump, 0, i, jump, 0, A], 0),
Arrows :=0, Width :=8,
LineStyle :=dashed, Ligne([Re(A), A, i*Im(A)], 0), LineStyle :=solid,
LabelDot(0, "$O$", "SE"), LabelDot(A, "$M$", "NO"),
LabelDot(i*Im(A), "$y$", "E"), LabelDot(Re(A), "$x$", "S"),
LabelDot(3/2, "$\vec{u}$", "S"), LabelDot(i/2, "$\vec{v}$", "E"),
LabelDot(3*A/4, "$\vec{w}=\vec{OM}$", "E")
] ;
\end{texgraph}

Figure 67

\begin{texgraph}[name=figure67, file]
Include "draw2d.mod" ;
Graph image = [
view(-3.5, 3.5, -2.5, 1.5), Marges(0, 0, 0, 0), size(8),
NewMac("f", "sin(x)", x),
Arrows :=1, Axes(0, 1+i, 2+2*i),
Width :=8, Color :=255, Arrows :=0, tMin :=-3, tMax :=3,
Cartesienne(f(x), 5), Color :=7709302,
draw("line", [y :=-3, ep :=0.5,

while y<3-ep do x :=y, Inc(y, ep), a :=f(x), x+i*a, y+i*a, jump, od],
[marker :=[0, Carc, 1, Oarc]]),

draw("line", [x :=2.5, a :=f(x), x+i*a, 3+i*a], marker :=[0, Carc, 1, Carc]),
Width :=8, LineStyle :=userdash, Color :=black,
x :=-3, ep :=0.5,
Ligne( [while x<3 do x, x+i*f(x), jump, Inc(x, ep) od, 3, 3+i*f(2.5)], 0),
Width :=8, LineStyle :=0, Color :=blue,
Courbe((0.5+f(t))*i+t), Color :=gray, Courbe((-0.5+f(t))*i+t),
Color :=0, tMin :=-3.5, tMax :=3.5, Color :=blue, LabelStyle :=left,
x :=0.8,
Ligne([1-x*i, 1.5-x*i], 0), Label(1.65-x*i, "$\mathcal{C}_{f+\varepsilon}$"),
x :=1.2,
Color :=red, Ligne([1-x*i, 1.5-x*i], 0), Label(1.65-x*i, "$\mathcal{C}_f$"),
Color :=gray, x :=1.6,
Ligne([1-x*i, 1.5-x*i], 0), Label(1.65-x*i, "$\mathcal{C}_{f-\varepsilon}$"),
Color :=Rgb(0.46, 0.63, 0.46), x :=2,
Ligne([1-x*i, 1.5-x*i], 0), Label(1.65-x*i, "$\mathcal{C}_{\psi}$")
] ;
\end{texgraph}

Figure 68

\begin{texgraph}[name=figure68, file]
Include "variations.mod" ;
Graph image = [
InitLig1("x", "-\infty", "0", "+\infty"),
InitCol1(+1, "\th", +1),
variationsLig(4, "-1",

+1,
["0", labelstyle :=framed, linestyle :=dotted, arrows :=0],
+1,
"+1"

),
Width :=8, traitH(1), traitV(1)
] ;
\end{texgraph}

Figure 69

\begin{texgraph}[name=figure69, file]
Cmd [view(-2.5, 3.5, -1, 2.5), Marges(0, 0, 0, 0), size(8)] ;
Mac f = exp(sin(%1))-1 ;

rectangles = [pas :=(%3-%2)/%4, u :=%2,
for k from 1 to %4 do

Assign(%1, t, u), y :=%1,
u, u+i*y, Inc(u, pas), u+i*y, u, jump

od
] ;

Graph objet = [Arrows :=1,
Axes(0, 1+i, 2+2*i), Width :=8, Arrows :=0, FillStyle :=bdiag, FillColor :=black,
n := 10, Ligne( rectangles(\f(t), -2, 3, n), 1),
Width :=14, Color :=red, FillStyle :=none, tMin :=-2, tMax :=3,
Courbe(t+i*\f(t), 2)
] ;

\end{texgraph}

Figure 70

\begin{texgraph}[name=figure70, file]
Graph image = [
view(-1, 5, -1, 5), Marges(0, 0, 0, 0), size(8, 1),
Arrows :=1, Axes(0, 1+i), Arrows :=0,
LineStyle :=dashed, Ligne([i, 1+i, 1], 0), LineStyle :=solid,
Width :=12,
Ddroite(0, 1+i), LabelDot(4+4*i, "$\alpha=1$", "E"),
tMin :=0,
Cartesienne(1/x), LabelDot(0.25+4.75*i, "$\alpha<0$", "E"),
Cartesienne(x^2), LabelDot(2+4*i, "$\alpha>1$", "E"),
Cartesienne(sqrt(x)), LabelDot(4+2*i, "$0<\alpha<1$", "SE")
] ;
\end{texgraph}

38 / 86



TeXgraph Exemples de graphiques

p(x) ker(s − id)

ker(p − id)

x −p(x)

ker(p) = ker(s + id)

x

s(x)

p(x)−x

Figure 71

\begin{texgraph}[name=figure71, file]
Graph image = [
view(-2, 6, -4, 5), Marges(0, 0, 0, 0), size(8), v :=1+3*i, x :=4+3*i,
p1 :=projO(x, [0, 1], v), p2 :=projO(x, [0, v], 1), y :=p1-p2,
Droite(0, 1), Droite(0, v),
Arrows :=1, Width :=8, Ligne([0, x, jump, 0, p1, jump, 0, p2, jump, 0, y, jump, 0, -p2], 0),
Arrows :=0,
Width :=4, Ligne([p2, x, y, -p2], 0), markseg(p1, x, 2, 0.15, 0.25), markseg(y, p1, 2, 0.15, 0.25),
LabelDot( p1/2, "$p(x)$", "S", 0, 0.15), LabelDot( Xmax, "$\ker(s-\id)$", "SO"),
LabelDot( Xmax, "$\ker(p-\id)$", "NO"),
LabelDot( p2/2, "$x-p(x)$", "O"), LabelDot(projO(i*Ymax, [0, v], 1), "$\ker(p)=\ker(s+\id)$",
"SE"),
LabelDot( x, "$x$", "NE"), LabelDot( y, "$s(x)$", "SE"), LabelDot( -p2/2, "$p(x)-x$", "O")
] ;
\end{texgraph}

x −∞ 0 +∞

ch

+∞

1

Figure 72

\begin{texgraph}[name=figure72, file]
Include "variations.mod" ;
Graph image = [
InitLig1("x", "-\infty", "0", "+\infty"),
InitCol1(+1, "\ch", +1),
variationsLig(2, "+\infty",

-2,
"1", +2,
"+\infty"

),
Width :=8, traitH(1), traitV(1)
] ;
\end{texgraph}

Figure 73

\begin{texgraph}[name=figure73, file]
Graph image = [
view(-3, 3, -2, 4), Marges(0, 0, 0, 0), size(6, 1),
Axes(0, 0), A :=2+3*i, d :=-2, H :=proj(A, [d, d+i]),
Arrows :=1, Width :=8, Ligne([0, 1, jump, 0, i], 0), Width :=4,
LabelArc(1, 0, A, 1.5, 1, "$t$"), Arrows :=0,
Ligne([H, A, 0], 0), Droite(d, d+i), DotScale :=1.5,
LabelDot(A, "$M$", "E", 1),
LabelAxe(x, 0.5, "$\vec{\imath}$", 1), LabelAxe(y, 0.5*i, "$\vec{\jmath}$", 2),
LabelDot(d, "$-d$", "SO"), LabelDot(0, "$F$", "SO"),
LabelDot(H, "$H$", "O", 1), LabelDot(d+Ymin*i, "$\mathcal{D}$", "NO")
] ;
\end{texgraph}

Figure 74

\begin{texgraph}[name=figure74, file]
Include "variations.mod" ;
Graph image = [
InitLig1("x", "-1", "0", "+1"),
InitCol1(+1, "\arcsin\,\,", +1),
variationsLig(4, "-\frac{\pi}2",

+1,
["0", labelstyle :=framed, linestyle :=dotted, arrows :=0],
+1,
"+\frac{\pi}2"

),
Width :=8, traitH(1), traitV(1)
] ;
\end{texgraph}

Figure 75

\begin{texgraph}[name=figure75, file]
Graph image = [
view(-1, 7.5, -1, 6.5), Marges(0, 0, 0, 0), size(8, 1),
Axes(0, 0), A :=4+3.5*i, Droite(0, A), u :=2*A/abs(A),
Arrows :=1, Width :=8, Ligne([0, 2, jump, 0, u, jump, 0, 2*i], 0),
Width :=6, Ligne([0, A], 0),
Width :=4, LabelStyle :=left, LabelArc(1, 0, A, 3, 1, "$\theta(t)$"),
LabelStyle :=center, Arrows :=0,
LabelAxe(x, 1, "$\vec{\imath}$", 1), LabelAxe(y, i, "$\vec{\jmath}$", 2),
LabelDot(u*0.5, "$\vec{u}_{\theta(t)}$", "E", 0, 0.35),
DotStyle :=cross, DotScale :=1.5, LabelDot(A, "$M(t)$", "E", 1),
A1 :=1+4.5*i, A2 :=3+5*i, A3 :=5+2*i, A4 :=6.5+2.5*i,
Spline(0, A1, A2, A, A3, A4, 0), LabelDot(A4, "$(\mathcal C)$", "N"),
LabelDot(A2, "$\vec{OM(t)}=\rho(t)\vec{u}_{\theta(t)}$", "N", 0, 0.5)
] ;
\end{texgraph}

Figure 76

\begin{texgraph}[name=figure76, file]
Graph image = [
view(-3, 3, -1.5, 1.5), Marges(0, 0, 0, 0), size(6, 0),
Arrows :=1, Axes(0, 1, 2), Arrows :=0,
LabelDot(i, "$1$", "NE"), LabelDot(-i, "$-1$", "SE"),
LineStyle :=dashed, Droite(0, 1, 1), Droite(0, 1, -1), LineStyle :=solid,
Width :=8, Color :=red, Cartesienne(th(x)),
LabelDot(1+th(1)*i, "$\mathcal{C}_{\th}$", "SE"),
Color :=black, Arrows :=2, tangente(th(x), 0, 2.5)
] ;
\end{texgraph}

Figure 77

\begin{texgraph}[name=figure77, file]
Cmd [view(-2.5, 3.5, -1.5, 2.5), Marges(0, 0, 0, 0), size(8)] ;
Mac f = sin(%1)+0.5 ;
Graph objet1 = [LineStyle :=noline, FillStyle :=bdiag, FillColor :=black,

domaine1(sin(t)+0.5, -2, 3), LineStyle :=dotted, FillStyle :=none, Width :=8,
Ligne( [-2, -2+f(-2)*i, jump, 3, 3+i*f(3)], 0),
Width :=14, Color :=red, LineStyle :=solid, tMin :=-2, tMax :=3,
Courbe(t+i*(sin(t)+0.5), 5), Width :=4, Color :=black, Arrows :=1,
Axes(0, 1+i, 2+2*i)
] ;

\end{texgraph}

Figure 78

\begin{texgraph}[name=figure78, file]
Cmd [view(-2.5, 3.5, -1, 2.5), Marges(0, 0, 0, 0), size(8)] ;
Mac f = exp(sin(%1))-1 ;

rectangles = [pas :=(%3-%2)/%4, u :=%2,
for k from 1 to %4 do

Assign(%1, t, u+pas/2), y :=%1,
u, u+i*y, Inc(u, pas), u+i*y, u, jump

od
] ;

Graph objet = [Arrows :=1,
Axes(0, 1+i, 2+2*i), Width :=8, Arrows :=0, FillStyle :=bdiag, FillColor :=black,
n := 10, Ligne( rectangles(\f(t), -2, 3, n), 1),
Width :=14, Color :=red, FillStyle :=none, tMin :=-2, tMax :=3,
Courbe(t+i*\f(t), 2)
] ;

\end{texgraph}

Figure 79

\begin{texgraph}[name=figure79, file]
Graph image = [
view(-3, 5, -3, 5), view3D(-1, 5, -1, 5, -1, 5), Marges(0, 0, 0, 0), size(8, 1),
theta :=20*deg, phi :=60*deg,
Ligne3D([ M(-1, 0, 0), M(5, 0, 0), jump, M(0, -1, 0), M(0, 5, 0), jump, M(0, 0, -1), M(0, 0, 5)], 0),
A :=M(3, 3, 4), H :=pxy(A), O :=Origin,
Arrows :=1, Width :=12,
Ligne3D([O, M(1, 0, 0), jump, O, M(0, 1, 0), jump, O, M(0, 0, 1), jump, O, A], 0),
Width :=4, Arc3D(M(1, 0, 0), O, H, 1.5, 1), Arrows :=0,
Ligne3D([O, H], 0), LineStyle :=dashed,
Ligne3D([px(A), H, A, pz(A), jump, H, py(A)], 0),
LabelSize :=footnotesize,
LabelDot3D(py(A), "$y$", "N", 1), LabelDot3D(px(A), "$x$", "O", 1),
LabelDot3D(pz(A), "$z$", "O", 1), LabelDot3D(A, "$A$", "E"), LabelDot3D(H, "$H$", "S"),
LabelDot3D([1.5*exp(i*angle(vecI, H)/2), 0], "$\theta$", "S"),
LabelDot3D(vecI, "$\vec{u}$", "O"), LabelDot3D(vecJ, "$\vec{v}$", "N"),
LabelDot3D(vecK, "$\vec{w}$", "O")
] ;
\end{texgraph}

C

θ0
θ

O

I

M

Figure 80

\begin{texgraph}[name=figure80, file]
Graph image = [
view(-1.5, 7, -2.5, 6), Marges(0, 0, 0.25, 0.25), size(6),
I :=3+2*i, R :=abs(I), DotStyle :=cross, DotScale :=1.5,
A :=I+R*exp(i*2*pi/3),
Width :=8, Arrows :=1, Axes(0, 0), Arrows :=0,
Ligne([A, 0, I], 0), Color :=red, Cercle(I, R),
LabelDot(I+R*exp(i*pi/4), "$\mathcal{C}$", "NE"), Color :=black,
Arrows :=1, LabelArc(1, 0, I, 1.75, 1, "$\theta_0$"),
LabelArc(1, 0, A, 1, 1, "$\theta$"),
Arrows :=0,
LabelDot(0, "$O$", "SO"), LabelDot(I, "$I$", "N", 1),
LabelDot(A, "$M$", "N", 1)
] ;
\end{texgraph}

Figure 81

\begin{texgraph}[name=figure81, file]
Graph image = [
view(-3, 4, -2, 4), view3D(-1, 2, -1, 3, -1, 2.5), Marges(0, 0, 0, 0), size(8, 1),
theta :=62*deg, phi :=43*deg,
$u :=vecI, $v :=[-1+i/2, 0], $w :=[(1+i)/2, 2.5], $M :=M(1.5, 2, 2),
SetMatrix3D([0, 0, u, v, w]),
Width :=4, Axes3D(0, 0, 0),
Width :=12, Arrows :=1, Ligne3D([0, 0, vecI, jump, 0, 0, vecJ, jump, 0, 0, vecK], 0), Arrows :=0,
Width :=4, Color :=gray, Dparallelep(Origin, 1.5*vecI, 2*vecJ, 2*vecK, 1), Color :=black,
Width :=8, LineStyle :=solid, Arrows :=1, Ligne3D([0, 0, M], 0), Arrows :=0,
DotStyle :=cross,
LabelDot(Proj3D(M), "$M$", "N", 1),
LabelDot(Proj3D(vecI), "$\vec{u}$", "N"),
LabelDot(Proj3D(vecJ), "$\vec{v}$", "N"),
LabelDot(Proj3D(vecK), "$\vec{w}$", "O"),
LabelDot(Proj3D([0, 0]), "$O$", "SE"),
LabelDot(Proj3D(0.85*M), "$\vec{a}=\vec{OM}$", "O")
] ;
\end{texgraph}

M

A
H

D

Figure 82

\begin{texgraph}[name=figure82, file]
Graph image = [
view(-3.5, 7.5, -3, 6), Marges(0, 0, 0, 0), size(6, 1),
theta :=-5*deg, phi :=74*deg, A :=M(0, 0, 5), B :=M(3, 3, 0),
Ligne3D([A, Origin, B], 1),
Droite(Proj3D([Origin, B])),
angleD(Proj3D(A), 0, Proj3D(B), 0.25),
LabelDot3D(A, "$M$", "N", 1), LabelDot3D(B, "$A$", "NE", 1),
LabelDot(0, "$H$", "NO", 1),
LabelDot3D(B/2, "$\mathcal{D}$", "S")
] ;
\end{texgraph}

Figure 83

\begin{texgraph}[name=figure83, file]
Graph image = [
Marges(0, 0, 0, 0),
O :=0, O' :=5.5, R :=4, R' :=1.5, a :=abs(O'-O), d :=(R^2-R'^2+a^2)/(2*a),
view(-0.5-R, O'+0.5+R', -0.5-R, 0.5+R), size(8),
Width :=6, Cercle(O, R), Cercle(O', R'), Droite(d, d+i), Droite(O, O'),
Arrows :=2, Ligne([O+i*Ymin, O'+i*Ymin], 0), Arrows :=0,
LineStyle :=dashed, Ligne( [O, O+i*Ymin, jump, O'+i*Ymin, O'], 0),
LineStyle :=solid, DotStyle :=cross,
LabelDot(O, "$O$", "N", 1), LabelDot(O', "$O'$", "N", 1),
SaveAttr(), FillStyle :=full, LineStyle :=noline, LabelStyle :=framed,
Label( (O+O')/2+i*Ymin, "$a$" ), RestoreAttr(),
LabelDot( O+R*i, "$\mathcal{C}$", "NO", 0),
LabelDot( O'+R'*i, "$\mathcal{C'}$", "NE", 0),
LabelDot( d+R*i, "$\mathcal{D}$", "NE", 0)
] ;
\end{texgraph}

Figure 84

\begin{texgraph}[name=figure84, file]
Graph image = [
view(-1, 4, -1, 3), Marges(0, 0, 0, 0), size(8+8*i, 0), DotStyle :=cross,
Axes(0, 0), A :=3.5+2.5*i, B :=1.5+1.75*i,
LineStyle :=dashed, for z in [A, B] do Ligne([i*Im(z), z, Re(z)], 0) od, LineStyle :=solid,
Arrows :=1, Width :=8,
Ligne([0, 1, jump, 0, i, jump, 0, A, jump, 0, B, jump, B, A], 0),
Arrows :=0,
LabelAxe(x, 0.5, "$\vec{\imath}$"), LabelAxe(x, Re(B), "$a$"),
LabelAxe(y, i*0.5, "$\vec{\jmath}$"),
LabelAxe(x, Re(A), "$x(t)$", 1, 1), LabelAxe(x, Re(B), "$a$", 1, 1),
LabelAxe(y, i*Im(A), "$y(t)$", 2, 1), LabelAxe(y, i*Im(B), "$b$", 2, 1),
LabelDot(0, "$O$", "SO"),
LabelDot(A/2, "$\vec{f}(t)$", "E", 0, 0.35),
LabelDot(B*0.75, "$\vec{\ell}$", "O", 0),
LabelDot(B+(A-B)*0.5, "$\vec{f}(t)-\vec{\ell}$", "O", 0, 0.45)
] ;
\end{texgraph}

Figure 85

\begin{texgraph}[name=figure85, file]
Cmd [view(-2.5, 3.5, -1, 2.5), Marges(0, 0, 0, 0), size(8)] ;
Mac f = exp(sin(%1))-1 ;

rectangles = [pas :=(%3-%2)/%4, u :=%2,
for k from 1 to %4 do

Assign(%1, t, u+pas), y :=%1,
u, u+i*y, Inc(u, pas), u+i*y, u, jump

od
] ;

Graph objet = [Arrows :=1,
Axes(0, 1+i, 2+2*i), Width :=8, Arrows :=0, FillStyle :=bdiag, FillColor :=black,
n := 10, Ligne( rectangles(\f(t), -2, 3, n), 1),
Width :=14, Color :=red, FillStyle :=none, tMin :=-2, tMax :=3,
Courbe(t+i*\f(t), 2)
] ;

\end{texgraph}

Figure 86

\begin{texgraph}[name=figure86, file]
Include "variations.mod" ;
Graph image = [
InitLig1("x", "-\infty", "0", "+\infty"),
InitCol1(+1, "\sh", +1),
variationsLig(4, "-\infty",

+1,
["0", labelstyle :=framed, linestyle :=dotted, arrows :=0],
+1,
"+\infty"

),
Width :=8, traitH(1), traitV(1)
] ;
\end{texgraph}

Figure 87

\begin{texgraph}[name=figure87, file]
Graph image = [
view(-6.5, 6.5, -5.25, 5.75), Marges(0, 0, 0, 0), size(8, 1),
theta :=33*deg, phi :=74*deg,
S :=Sphere(Origin, 4, 25, 25), d :=-2, x :=sqrt(16-d^2), H :=M(0, 0, d),
P :=[H, vecK], M :=[x*exp(i*80*deg), d],
Build3D(

bdPlan( P, [color := darkseagreen]),
bdFacet(S, [color :=steelblue, backculling :=1, smooth :=1]),
bdCercle([0, 0], 4.01, vecK, [width :=8]),
bdAxes([0, 0], [labels :=0, width :=12])
),

Display3D(),
LineStyle :=dotted, Width :=12, Arrows :=1, Axes3D(0, 0, 0), Arrows :=0,
Width :=8, Cercle3D([0, 0], 4.01, vecK),
Ligne3D([Origin, M, H], 0),
LineStyle :=solid, angleD(0, Proj3D(H), Proj3D(M), 0.25),
DotStyle :=bigdot,
LabelDot( Proj3D(H), "$H$", "O", 1),
LabelDot( 0, "$O$", "NO", 1),
I :=Intersection( P, S), Color :=red, DrawAretes(I, 0),
Color :=black, LabelDot( Proj3D(M), "$M$", "E", 1)

] ;
\end{texgraph}

a b

A

B

c1 c2

Figure 88

\begin{texgraph}[name=figure88, file]
Graph image = [
view(-4, 4.5, -0.75, 8), Marges(0, 0, 0, 0), size(8+5*i, 0), a :=-3, b :=3.5,
Droite(1, 0), Droite(1, 0, a), Droite(1, 0, b),
LabelDot(a, "$a$", "SO", 1), LabelDot(b, "$b$", "SE", 1),
NewMac("f", "x+sin(x)*1.5+4", x),
tMin :=a, tMax :=b, Width :=8, Color :=red, Cartesienne(f(x)),
Color :=black, Width :=4, A :=a+i*f(a), B :=b+i*f(b),
Droite(A, B),
LabelDot(A, "$A$", "NO", 1), LabelDot(B, "$B$", "SE", 1),
T :=(f(b)-f(a))/(b-a), x :=Nil, L :=Solve(1+2*cos(x)-0.95, x, a, b),
Color :=darkblue, k :=1,
for x0 in L do

LineStyle :=solid, Width :=8, tangente(f(x), x0),
Width :=4, LineStyle :=dashed, Ligne([x0, x0+i*f(x0)], 0),
LabelDot(x0, ["$c_", k, "$"], "S", 1), Inc(k, 1)

od
] ;
\end{texgraph}

Figure 89

\begin{texgraph}[name=figure89, file]
Graph image = [
view(-3, 3, -6, 6), Marges(0, 0, 0, 0), size(6, 0),
Arrows :=1, Axes(0, 1+i), Arrows :=0,
Width :=8, Color :=red, Cartesienne(sh(x)),
LabelDot(2+sh(2)*i, "$\mathcal{C}_{\sh}$", "SE"),
Color :=black, Arrows :=2, tangente(sh(x), 0, 2.5)
] ;
\end{texgraph}

Figure 90

\begin{texgraph}[name=figure90, file]
Graph image = [
view(-5, 5, -5, 5), Marges(0, 0, 0, 0), size(6),
Width :=8, A :=-4+4*i, C :=4-2*i, H :=proj(C, [A, A-i]),
Droite(A, A-i), Ligne([A, C, H], 0), angleD(C, H, A, 0.25),
LabelDot(C, "$M$", "E", 1), LabelDot(A, "$A$", "O", 1),
LabelDot(H, "$H$", "O", 1), LabelDot(-4+4.5*i, "$\mathcal{D}$", "E")
] ;
\end{texgraph}

Figure 91

\begin{texgraph}[name=figure91, file]
Graph image = [
view(-3, 3, -3, 3), Width :=8, size(6),
A :=setB("A", -1, [scale :=0.5, rotation :=15]),
B :=setB("B", 1, [scale :=0.5, rotation :=-15]),
C :=setminusB(A, B),
FillColor :=lightgray, FillStyle :=full, Color :=red,
Bezier(C), LabelStyle :=framed,
Label(-2.5*i, "$A\setminus B$")
] ;
\end{texgraph}

Figure 92

\begin{texgraph}[name=figure92, file]
Graph image = [
view(-1, 5, -1, 4), Marges(0, 0, 0, 0), size(8), v :=1+3*i, x :=4+3*i,
p1 :=projO(x, [0, 1], v), p2 :=projO(x, [0, v], 1),
Droite(0, 1), Droite(0, v),
Arrows :=1, Width :=8, Ligne([0, x, jump, 0, p1, jump, 0, p2], 0), Arrows :=0,
Width :=4, Ligne([p2, x, p1], 0),
LabelDot( p1/2, "$p(x)$", "S"), LabelDot( Xmax, "$\ker(p-\id)$", "SO"),
LabelDot( Xmax, "$\im(p)$", "NO"),
LabelDot( p2/2, "$x-p(x)$", "O"),
LabelDot(projO(i*Ymax, [0, v], 1) , "$\ker(p)$", "SE"),
LabelDot( x, "$x$", "NE")
] ;
\end{texgraph}

Figure 93

\begin{texgraph}[name=figure93, file]
Graph image = [
view(-6, 6, -6, 6), Marges(0, 0, 0, 0), size(8),
Width :=4, Cercle(0, 5),
axes(0, 0, 0, 0), alpha :=40*deg, r :=-1.25,
u1 :=5*exp(i*alpha), u2 :=i*u1, A :=r*u1,
Width :=8, Arrows :=1, Ligne( [0, 5, jump, 0, 5*i, jump, 0, A], 0),
Width :=12, Ligne([0, u1, jump, 0, u2], 0),
Width :=8,
LabelArc(1, 0, u1, 2, 1, "$\theta$"), Arrows :=0,
LabelDot(0, "$O$", [0.75, exp(i*165*deg)]), LabelDot(u1*0.85, "$\vec{u}(\theta)$", "O"),
LabelDot(u2*0.75, "$\vec{v}(\theta)$", "E"), LabelDot(2.5, "$\vec{u}$", "S"),
LabelDot(2.5*i, "$\vec{v}$", "E"), LabelDot(5, "$1$", "SE"),
LabelDot(A, "$M$", "N"), LabelDot(-5.5-5.5*i, "$\vec{OM}=r\vec{u}(\theta)$", "E"),
angleD(u1, 0, u2, 0.25),
Width :=4, LineStyle :=dashed, Droite(0, u1), Droite(0, u2)
] ;
\end{texgraph}

Figure 94

\begin{texgraph}[name=figure94, file]
Include "variations.mod" ;
Graph image = [
InitLig1("x", "-1", "0", "+1"),
InitCol1(+1, "\argth\,\,", +1),
variationsLig(4, "-\infty",

+1,
["0", labelstyle :=framed, linestyle :=dotted, arrows :=0],
+1,
"+\infty"

),
Width :=8, traitH(1), traitV(1)
] ;
\end{texgraph}

Figure 95

\begin{texgraph}[name=figure95, file]
Graph image = [
view(-3, 3, -2.5, 2.5), Marges(0, 0, 0, 0), size(6),
Width :=6, Cercle(0, 2), Droite(2.5, 2.5+i),
Arrows :=2, Ligne([0, 2.5], 0), Arrows :=0, DotStyle :=cross,
LabelDot(0, "$O$", "N", 1),
LabelDot(2.5/2, "$d$", "S", 0),
LabelDot( 2*i, "$\mathcal{C}$", "NO", 0),
LabelDot( 2.5+2*i, "$\mathcal{D}$", "NE", 0),
Label(-2.25*i, "$d>R$")
] ;
\end{texgraph}

Figure 96

\begin{texgraph}[name=figure96, file]
Graph image = [
view(-3.5, 4.5, -3, 4), view3D(-3.5, 3.5, -3.5, 3.5, -3.5, 3.5), Marges(0, 0, 0, 0),
A :=M(3, 3, 3), Width :=8, Arrows :=1, Axes3D(0, 0, 0),
LabelDot3D(Origin, "$O$", "NO", 1),
Arc3D(px(A), Origin, pxy(A), 1.5, 1), Arc3D(pz(A), Origin, A, 1.5, 1),
Arrows :=0, LineStyle :=userdash,
Ligne3D([px(A), pxy(A), py(A), jump, A, pxy(A), Origin], 0),
Arrows :=1, LineStyle :=solid, Ligne3D([Origin , A], 0),
LabelDot3D(A, "$\vec{n}$", "NE"),
Label(0.1228-1.0377*i, "$\theta$"),
Label(0.3509+1.3396*i, "$\varphi$")
] ;
\end{texgraph}

Figure 97

\begin{texgraph}[name=figure97, file]
Graph image = [
view(-5, 4, -5, 5), Marges(0.5, 0, 0, 0.5),
size(6), Width :=2,
Color :=lightgray, Grille(-5-5*i, (1+i)/2),
Width :=4,
Color :=gray, Grille(-5-5*i, (1+i)),
Color :=black, Arrows :=1,
Axes(-5-5*i, 1+i, 1+i), Arrows :=0,
LabelAxe(x, -pi-5*i, "$-\pi$", 2-i, 1),
LabelAxe(x, pi-5*i, "$\pi$", 2+i, 1),
SaveAttr(),
FillStyle :=full, FillColor :=lightblue,
FillOpacity :=0.5,
domaine2(3*sin(x)^3, -5, -pi, pi),
RestoreAttr(),
Color :=red, Arrows :=0, Width :=8,
Cartesienne( 3*sin(x)^3 )
] ;
\end{texgraph}

Figure 98

\begin{texgraph}[name=figure98, file]
Graph image = [
view(-5, 5, -4, 6), Marges(0, 0, 0, 0), size(6),
Axes(2*i, 1+i), Eofill :=1,
FillStyle :=full, FillOpacity :=0.9,
FillColor := blue, Width :=8,
Path([-4, i, circle,

-3+2*i, move, -3, -2, line,
0, 2, 2, -1, arc,
3, 3+3*i, 0.5, linearc,
1, -1+5*i, -3+2*i, bezier,
closepath,
])

] ;
\end{texgraph}

Figure 99

\begin{texgraph}[name=figure99, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
P := for k from 1 to 10 do

z :=8*Rand()-4+i*(8*Rand()-4),
LabelDot(z, k, "N", 1), z
od,

Ligne(Cvx2d(P), 1)
] ;
\end{texgraph}

R T F
Figure 100

\begin{texgraph}[name=figure100, file]
Graph image = [
view(-5.25, 5.25, -5.25, 5.25),
Marges(0, 0, 0, 0), size(7.5),
background(full, blue),
Width :=4, Color :=white,
inclin :=[0, 35, -35],
for z in inclin do
Ellipse(0, 5, 2, z)

od,
Width :=2*mm, Ellipse(0, 1.5, 4.5),
Label(-0.1,

"\resizebox{6cm}{3.5cm}{R\ T\ F}")
] ;
\end{texgraph}

Figure 101

\begin{texgraph}[name=figure101, file]
Graph image = [
view(-10, 10, -5, 5), size(7.5), NbPoints :=100,
LabelSize :=footnotesize, SaveWin(), view(-10, -1, -5, 5),
ChangeWinTo([-2-2*i, 2+2*i]),
Arrows :=1, axes(0, 1+i), Arrows :=0,
tMin :=-2, tMax :=2, Color :=red, Width :=8,
Cartesienne(x*Ent(1/x)-1, 5, 1),
Color :=blue, A :=(1+i)/4, Dparallelo(A, bar(A), -A),
dep :=RealCoord(i*Im(A)), RestoreWin(), SaveWin(),
//zoom
view(1, 10, -5, 5), background(full, white),
ChangeWinTo([-A, A]), Color :=black,
arr :=RealCoord(-Re(A)+i*Im(A)*0.75),
Arrows :=1, axes(0, A), Arrows :=0,
tMin :=-0.25, tMax :=0.25, Color :=red, Width :=8,
Cartesienne(x*Ent(1/x)-1, 5, 1),
Color :=blue, Dparallelo(A, bar(A), -A),
RestoreWin(),
//trait
Color :=blue, Arrows :=1,
A :=ScrCoord(dep), B :=ScrCoord(arr),
Bezier(A, A+3*exp(i*pi/2), B-3, B)
] ;
\end{texgraph}

Figure 102

\begin{texgraph}[name=figure102, file]
Graph image = [
Marges(0, 0.15, 0, 0), size(7.5),
C :=rotCurve(
2*[1.5*i+cos(3*t)*cos(t)*i, -cos(3*t)*sin(t)],
[0, 0, 0, 1], -pi/2, pi/2, 0, pi, 50),

DrawFlatFacet(C, [color :=coral]),
tMin :=-pi/2, tMax :=pi/2, Color :=blue,
Width :=8,
Courbe3D(M(0, 3+2*cos(3*t)*cos(t),

2*cos(3*t)*sin(t)))
] ;
\end{texgraph}

Figure 103

\begin{texgraph}[name=figure103, file]
Graph image = [
view(-5, 5, -5, 5), Marges(0.25, 0.25, 0.25, 0.25),
size(6), Width :=4, Axes(0, 1+i),
A := -4-3*i, B :=-2+2*i, C :=1-3*i, D :=4+3*i,
LabelDot(A, "$A$", "S", 1), LabelDot(B, "$B$", "N", 1),
LabelDot(C, "$C$", "S", 1), LabelDot(D, "$D$", "O", 1),
Width :=8, Color :=red, Spline(0, A, B, C, D, 0),
Ligne([-4.5+4.5*i, -4+4.5*i], 0), LabelStyle :=left,
Label(-3.5+4.5*i, "libre"),
Color :=blue, Spline(5, A, B, C, D, 5*i),
Ligne([-4.5+3.5*i, -4+3.5*i], 0),
Label(-3.5+3.5*i, "contrainte"),
Width :=4,
Arrows :=1, Ligne([A, A+2, jump, D, D+2*i], 0)
] ;
\end{texgraph}

Figure 104

\begin{texgraph}[name=figure104, file]
Cmd Fenetre(-5+5*i, 5-5*i, 0.5+0.5*i) ;

Marges(0, 0, 0, 0) ;
[theta :=1*deg, phi :=83*deg] ;

Graph objet1 = [
C1 :=Cylindre(M(-4, 0, 0), 8*vecI, 2, 25),
C2 := Cylindre(M(0, 0, -4), 8*vecK, 3, 25),
stock := Clip3D(C2, C1, 1),
DrawSmoothFacet(stock,

[color :=darkseagreen])
] ;
\end{texgraph}

Figure 105

\begin{texgraph}[name=figure105, file]
Graph image = [
view(-6.5, 6, -6.5, 5.5),
Marges(0, 0, 0, 0), size(6),
view3D(-3, 3, -3, 3, -3, 3), ModelView(central),
S :=GetSurface([u+i*v, 2*sin(u)+cos(v)],

-3+3*i, -3+3*i),
stock :=for facette in S By jump do

z :=Zde(isobar3d(facette)),
facette,
ColorJump(Hsb(270*(Zsup-z)/(Zsup-Zinf), 1, 1))

od,
FillStyle :=full, LabelSize :=footnotesize,
BoxAxes3D(grid :=1, FillColor :=lightblue),
Ligne3D(SortFacet(stock), 1)
] ;
\end{texgraph}

Figure 106

\begin{texgraph}[name=figure106, file]
Include "PolyedresII.mac" ;
Graph image = [
Marges(0, 0, 0, 0), size(6),
DrawFacet(Sphere(M(0, 0, 3.5), 1.5, 30, 15),

[color :=steelblue,
backculling :=1, smooth :=1],

Cube([0, 0, vecK], M(1, 1, 0)),
[color :=orange, smooth :=0],

Cube([0, 0, vecK], M(2, 2, -4)),
[color :=gold]
)

] ;
\end{texgraph}

Figure 107

\begin{texgraph}[name=figure107, file]
Graph image = [
Marges(0, 0, 0, 0), view(-3, 3, -3, 3),
view3D(-3, 3, -3, 3, -3, 3), size(6),
background(full, gray),
B :=M(0, 2, 0), A :=M(0, 0, 0), C :=M(0, 0, 1.5),
Build3D(
bdAngleD(B, A, C, 1, [color :=firebrick, tube :=1]),
bdDot([A, B, C], [dotstyle :=cube,

dotscale :=0.85,
color :=forestgreen]),

bdArc(B, A, C, 2, 1, [color :=blue, width :=12,
arrows :=1, labelarc("$\pi/2$")]),

bdAxes([0, 0], [color :=gold, arrows :=1]),
bdLabel(B, "$B$", [labelpos :=[0.5, -i]]),
bdLabel(C, "$C$", [labelpos :=[0.5, -1]]),

),
Display3D()
] ;
\end{texgraph}

Figure 108

\begin{texgraph}[name=figure108, file]
Graph image = [
view(-3, 2, -2, 3), Marges(0.25, 0.25, 0.25, 0.25),
size(6), Width :=4,
Axes(0, 1+i), NbPoints :=250, tMin :=-25, tMax :=25,
courbe :=Get(Polaire((t+1)/(t-1))),
ptDoubles := courbe InterL courbe,
Width :=8, Color := blue, Ligne(courbe, 0),
DotStyle :=dotcircle, DotScale :=2,
Point(ptDoubles),
Label(1+2*i, "$r(t)=\dfrac{t+1}{t-1}$")
] ;
\end{texgraph}

Figure 109

\begin{texgraph}[name=figure109, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
S :=curveTube([i*t, 2.5*sin(t)], 0.75, -pi, pi, 50, 12),
DrawFlatFacet(S, [color :=darkseagreen])
] ;
\end{texgraph}

Figure 110

\begin{texgraph}[name=figure110, file]
Graph image = [
Marges(0, 0, 0, 0), view(-3, 3, -3, 3),
view3D(-3, 3, -3, 3, -3, 3), size(6),
B :=M(0, 2, 0), A :=M(0, 0, 0), C :=M(0, 0, 1.5),
Build3D(

bdAxes([0, 0], [color :=gold, arrows :=1]),
bdPlan([0, 0, 1+i, 2], [color :=darkseagreen,

scale :=0.75]),
bdSurf(M(u, -v, sqrt(u^4+v^4)-2),

[color :=steelblue, u :=[-2, 2],
v :=u, smooth :=1, clip :=1,
clipwin :=[M(-3, -3, -3), M(3, 3, 2)]]),

bdLabel([0.25*(1+i), 2.25], "z=\sqrt{x^4+y^4}-2",
[TeXify :=1, scale :=0.75])

),
Display3D()
] ;
\end{texgraph}

Figure 111

\begin{texgraph}[name=figure111, file]
Mac

dessin = [ BoxAxes3D(grid :=1, zlabelstyle :=right,
zlabelsep :=0.15, xlabelsep :=0.25,

ylabelsep :=0.25,
xlegendsep :=0.35, ylegendsep :=0.35,
FillColor :=lightcyan),

Ligne3D(SortFacet(stock), 1), RestoreWin()] ;
Cmd [tMin :=-5, tMax :=0, DotScale :=1+i] ;
Graph objet1 = [view(-6, 6, -6, 6), Marges(0, 0, 0, 0),

size(7.5),
view3D(-3, 3, -3, 3, -3, 3),
S :=GetSurface([u+i*v, 2*sin(u)+cos(v)],

-3+3*i, -3+3*i),
stock :=for facette in S By jump do

z :=Zde(isobar3d(facette)),
facette,
ColorJump(Hsb(270*(Zsup-z)/(Zsup-Zinf), 1, 1))

od,
FillStyle :=full, LabelSize :=tiny,
ModelView(central), SaveTphi(), SaveWin(),
view(-6, 0, 0, 6), ChangeWinTo([-8-7*i, 6+6*i]),
dessin(), SaveWin(), ModelView(ortho),
view(0, 6, 0, 6), ChangeWinTo([-6-6*i, 4+5*i]),
viewDir(xOy), dessin(), SaveWin(),
view(-6, 0, -6, 0), ChangeWinTo([-6-6*i, 4+5*i]),
viewDir(yOz), dessin(), SaveWin(),
view(0, 6, -6, 0), ChangeWinTo([-6-6*i, 4+5*i]),
viewDir(xOz), dessin(), RestoreTphi() ] ;

\end{texgraph}

Figure 112

\begin{texgraph}[name=figure112, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
DrawSmoothFacet( Sphere(M(-3, 0, 0), 3, 25, 15),

[color :=steelblue,
backculling :=1],

Sphere(M(3, 0, 0), 3, 25, 15),
[color :=orange])

] ;
\end{texgraph}

Figure 113

\begin{texgraph}[name=figure113, file]
Graph image = [
view(-5, 5, -5, 5), size(6),
Axes(0, 1+i), Width :=8,
Color :=red,
periodic(if t>0 then 4 else 2 fi, -1, 1, 5, 1),
Color :=blue,
periodic(2*abs(t)-1, -1, 1),
Color :=magenta,
periodic(sin(t)-3, -pi/2, pi/2)
] ;
\end{texgraph}

Figure 114

\begin{texgraph}[name=figure114, file]
Graph image = [
view(-5, 5, -5, 5), size(7.5),
LabelSize :=footnotesize,
//graphique du haut
SaveWin(),
view(-5, 5, 0.25, 5), Width :=6, Arrows :=1,
SetMatrix([2.5*i, 2/pi, 2*i]),
axeX(0, pi/2, Nil, Nil, 1, "\pi", 2), axeY(0, 1),
Arrows :=0, Color :=red, Width :=8,
tMin :=-2*pi, tMax :=2*pi, Cartesienne(sin(x)),
Label(pi*1.65+i, "$f(x)=\sin(x)$"), RestoreWin(),
//graphique du bas
SaveWin(), Color :=black,
view(-5, 5, -5, -0.25), SetMatrix([-5.5*i, 2, i]),
Width :=6, Arrows :=1, usecomma :=1,
axeX([i, 0], 0.5), axeY([i, 0], 1, Nil, Nil, 1, "a"),
Arrows :=0, Color :=blue, Width :=8,
Cartesienne(x^2+1),
Label(1+4.75*i, "$g(x)=x^2+a$"),
RestoreWin()
] ;
\end{texgraph}

Figure 115

\begin{texgraph}[name=figure115, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
S :=GetSurface([u+i*v, sin(u)+cos(v)],

pi*(-1+i), pi*(-1+i)),
DrawFacet(S, [color :=Rgb(0.74, 0.73, 1)])
] ;
\end{texgraph}

Figure 116

\begin{texgraph}[name=figure116, file]
Graph image = [
view(-2, 3, -2, 4.5), Marges(0, 0, 0, 0), size(6),
plan :=[M(1.5, 0, 0), -vecI],
S := Section(plan,

Tetra(Origin, 3*vecI, 3*vecJ, 4*vecK)),
A :=Aretes(S), Width :=12, DrawAretes(A)

] ;
\end{texgraph}

A B

Figure 117

\begin{texgraph}[name=figure117, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
A :=setB("A", 0, [rotation :=30]),
B :=setB("B", 0, [rotation :=-30]),
C := capB(A, B), Color :=red, FillStyle :=full,
drawSet(C, [FillOpacity :=0.5, FillColor :=pink])
] ;
\end{texgraph}

Figure 118

\begin{texgraph}[name=figure118, file]
Graph image = [
view(-5, 5, -3, 3), size(6),
SetMatrix([0, 1, 1+i]), axes(0, 1+i),
tMin :=-5, tMax :=5,
Color :=red, Width :=8, Cartesienne(2*sin(x)),
Color :=black, Arrows :=2,
tangente(2*sin(x), pi/2, 1.5),
Arrows :=0, LineStyle :=dotted,
Ligne( [2*i, pi/2+2*i, pi/2], 0),
Point(pi/2+2*i),
LabelDot( pi/2, "$\frac{\pi}2$", "S", 1),
IdMatrix()
] ;
\end{texgraph}

Figure 119

\begin{texgraph}[name=figure119, file]
Graph image = [
view(-5, 5, -2.5, 2.5), Marges(0, 0, 0, 0), size(7.5),
F :=for z in carre(3-3*i, 3+3*i, 1) do z, 0 od,
theta :=-pi/2, phi :=0, ep :=2, SaveWin(),
view(-5, 0, -2.5, 2.5), ChangeWinTo([-5-5*i, 5+5*i]),
DrawFacet(HollowFacet(F, ep), [color :=lightblue]),
Arrows :=2, Ligne3D([-3+3.25*i, 0, (-3+ep/2)+3.25*i, 0], 0),
Arrows :=0, Label(-3+ep/4+4*i, "ep/2"),
RestoreWin(), SaveWin(),
view(0, 5, -2.5, 2.5), ChangeWinTo([-5-5*i, 5+5*i]),
DrawFacet(HollowFacet(F, ep+i), [color :=lightblue]),
Arrows :=2, Ligne3D([3+3.25*i, 0, (3-ep)+3.25*i, 0], 0),
Arrows :=0, Label(3-ep/2+4*i, "ep"),
RestoreWin(),
Label(-2.5-2*i, "mode=0"), Label(2.5-2*i, "mode=1")
] ;
\end{texgraph}

\begin{texgraph}[name=figure120, file]
Graph image = [
view(-5, 5, -5, 5), view3D(-4, 4, -4, 4, -4, 4),
size(7.5), plan :=[Origin, vecJ],
C :=for t from -2*pi to 2*pi step 0.1 do

[2*exp(i*t), t/3] od,
L :=Clip3DLine(C, plan, 0, L'),
Ligne3D([M(0, -4, 0), Origin], 0),
Color :=blue, Ligne3D(L', 0),
FillStyle :=full, FillColor :=gray,
FillOpacity :=0.8, Color :=black,
DrawPlan( [Origin, vecJ], vecI, 5, 5),
FillStyle :=none,
Yinf :=0, Axes3D(0, 0, 0),
Color :=red, Ligne3D(L, 0)

] ;
\end{texgraph}

Figure 121

\begin{texgraph}[name=figure121, file]
Cmd view(0, 6.25, -1.5, 2) ;

size(7.5) ;
[tMin :=0, tMax :=6.25] ;

Graph axe = Axes(0, 1+i) ;
C1 = [Color :=blue, Width :=8,

Cartesienne(1/x) ] ;
C2 = [Color :=red, Width :=8,

Cartesienne(sin(2*x)*1.5)] ;
inter = [DotScale :=2, DotStyle :=dotcircle,

Color :=forestgreen,
Point(Get(C1) InterL Get(C2)),
Color :=blue,
LabelDot(5.5+0.1*i, "$C_1$", "N"),
Color :=red,
LabelDot(3-i, "$C_2$", "E"),
] ;

\end{texgraph}

Figure 122

\begin{texgraph}[name=figure122, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
A :=-5-5*i, B :=5*i, C :=5-5*i, niv :=6,
Tr :=[A, B, C, jump], {initial}
for k from 1 to niv do
Tr :=[hom(Tr, A, 0.5), hom(Tr, B, 0.5),

hom(Tr, C, 0.5)]
od,
FillStyle :=full, FillColor :=blue,
Ligne(Tr, 1)
] ;
\end{texgraph}

Figure 123

\begin{texgraph}[name=figure123, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
theta :=75*deg, phi :=60*deg,
S :=Cylindre(M(0, 0, -4), 8*vecK, 2, 25, 0),
C := curveTube([3*exp(i*t), t/3],

0.5, -2*pi, 2*pi, 75, 12, 0),
DrawFlatFacet(S, [color :=steelblue,

backculling :=1],
C, [color :=crimson])

] ;
\end{texgraph}

Figure 124

\begin{texgraph}[name=figure124, file]
Graph image = [
ModelView(central), Marges(0, 0, 0, 0), size(6),
view3D(-3, 3, -3, 3, -3, 3),
S :=clipPoly( Sphere( M(1, 0, 1), 3, 30, 15) ),
HideWidth :=8, drawWin3d(0),
DrawFacet(S, [color :=steelblue]),
drawWin3d(1)
] ;
\end{texgraph}

Figure 125

\begin{texgraph}[name=figure125, file]
Graph image = [
view(-5, 5, -4, 5), size(7.5), tMin :=-5,
Axes(-5, 1+i, 2+2*i), tMax :=-1,
Cartesienne(sin(x)), tMax :=5,
Cartesienne(2*cos(x)), tMin :=-3, tMax :=3,
Cartesienne(x^2), A :=-4.5+(Ymin+1.75)*i,
FillStyle :=full, FillOpacity :=0.5, FillColor :=green,
LabelStyle :=left, LabelSize :=footnotesize,
Dcarre(A+0.25*(-1+i), A+0.25*(1+i), -1),
Label(A+0.4, "domaine2 sur $[-4\, ;\, -1.5]$ "),
Inc(A, -0.75*i), FillColor :=blue,
Dcarre(A+0.25*(-1+i), A+0.25*(1+i), -1),
Label(A+0.4, "domaine3 entre $2\cos(x)$ et $x^2$ "),
Inc(A, -0.75*i), FillColor :=red,
Dcarre(A+0.25*(-1+i), A+0.25*(1+i), -1),
Label(A+0.4, "domaine1 sur $[1.1\, ;\, 4]$"),
FillColor :=red, domaine1(2*cos(x), 1.1, 4),
FillColor :=green, domaine2(2*cos(x), sin(x), -4, -1.5),

FillColor :=blue, domaine3(2*cos(x), x^2),
Arrows :=2, tangente(2*cos(x), 0.5, 1.5)
] ;
\end{texgraph}

Figure 126

\begin{texgraph}[name=figure126, file]
Graph image = [
view(-0.1, 1.1, -0.1, 1.1), size(6),
Arrows :=1, Axes(0, 1/2+i/4), Arrows :=0,
u0 :=0.3, nb :=20, Width :=6,
Color :=darkseagreen,
Droite(1, -1, 0), Color :=red,
tMin :=0, tMax :=1,
Width :=8, Cartesienne(4*x*(1-x)),
Width :=6, Color :=blue,
Label(0.5+1.05*i, "$u_{n+1}=4u_n(1-u_n)$"),
LabelDot(u0, "$u_0$", "S", 1),
suite(4*x*(1-x), u0, nb)
] ;
\end{texgraph}

A B

Figure 127

\begin{texgraph}[name=figure127, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
A :=set("A", 0, [rotation :=30]),
B :=set("B", 0, [rotation :=-30]),
C := setminus(A, B), Color :=red, FillStyle :=full,
FillOpacity :=0.5, FillColor :=pink,
Ligne(C, 0)
] ;
\end{texgraph}

Figure 128

\begin{texgraph}[name=figure128, file]
Graph image = [
Marges(0, 0, 0, 0), size(6), NbPoints :=100,
C :=curve2Cone( [cos(2*t)*exp(i*t)*5, -3],

-pi, pi, [0, 0], -1),
FillOpacity :=0.7,
DrawFlatFacet(C, [color :=darkseagreen])
] ;
\end{texgraph}

y

z

x

Figure 129

\begin{texgraph}[name=figure129, file]
Cmd Marges(0, 0, 0, 0) ;

size(6) ;
Include "PolyedresII.mac" ;
Var A = M(3, 0, 0) ;
Mac f = M(%1, %2, sin(%1)+cos(%2)) ;
Graph objet1 = [

background(full, lightgray),
ColorL :=slategray, StyleL :=12,
CubeAdc(Origin, A, C, T1, T2, Ar),
Build3D(C, T1, T2, Ar,

bdSurf(f(u, v),
[color :=steelblue,
u :=[-pi, pi], v :=u,
contrast :=0.125]),

bdAxes(Origin,
[color :=forestgreen, arrows :=1])

),
Display3D()] ;

\end{texgraph}

Figure 130

\begin{texgraph}[name=figure130, file]
Graph image = [
view(-5, 5, -3, 6), Marges(0, 0, 0, 0), size(6),
viewDir(115, 70),
P :=shift3d(Pyramide( [Origin, M(1, -1, 0), M(3, 1, 0),

M(3, 3, 0), M(1, 4, 0)], M(1, 1, 3)), 2*vecJ),
miroir :=[M(-4, 0, 0), M(4, 0, 0), M(4, 0, 5), M(-4, 0, 5), jump],
P' :=reverse3d( Mtransform3D( P,

matrix3d(sym3d(M, [Origin, vecJ])))),
FillStyle :=full, FillColor :=brown, Width :=8,
DrawFacet( P', [color :=FillColor]),
DrawFacet( miroir,

[FillOpacity :=0.5, color :=lightgray]
),

DrawPoly(P, 4)
] ;
\end{texgraph}

0 1 2−1−2

1

2

Figure 131

\begin{texgraph}[name=figure131, file]
Graph image = [
view(-2, 2, -0.1, 2), Marges(0.5, 0.5, 0.5, 0.5),
size(6), tMin :=-2, tMax :=2,
Color :=darkgray, Width :=8,
LineStyle :=dotted, Grille(0, 0.5*(1+i)),
Color :=black, LineStyle :=solid, Axes(0, 1+i, 1),
NbPoints :=100, Width :=8, Color :=red,
Cartesienne(x*Ent(1/x), 5, 1)
] ;
\end{texgraph}

A B

Figure 132

\begin{texgraph}[name=figure132, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
A :=set("A", 0, [rotation :=30]),
B :=set("B", 0, [rotation :=-30]),
C := cup(A, B), Color :=red, FillStyle :=full,
FillOpacity :=0.5, FillColor :=pink,
Ligne(C, 0)
] ;
\end{texgraph}

Figure 133

\begin{texgraph}[name=figure133, file]
Graph image = [
view(-5, 5, -1, 3), Marges(0, 0, 0, 0),
size(7.5), Seg(-5, 5),
for t in [-4, -1.85, 0, 1.85, 3] do

M :=t-sin(t)+i*(1-cos(t)),
I :=t+i, DotStyle :=cross,
Point(I), DotStyle :=bigdot,
Point(M), Cercle(I, 1), Seg(M, I),
Arrows :=1, Arc(M, I, t, 0.5, t),
Arrows :=0, LineStyle :=dashed,
Seg(I, t), LineStyle :=solid

od,
Width :=8, Color :=red,
Courbe( t-sin(t)+i*(1-cos(t)))
] ;
\end{texgraph}

Figure 134

\begin{texgraph}[name=figure134, file]
Include "PolyedresII.mac" ;
Graph image = [
view(-5, 5, -5, 5), Marges(0, 0, 0, 0),
size(7.5, 0),
C :=Cube(Origin, M(3, 3, 0)),
S :=Sommets(C), Point3D(S),
DrawPoly(C, 0), k :=0,
for Z in S by 2 do
Inc(k, 1),
Label(Proj3D(Z)+
if k>4 then 0.5*i else -0.5*i fi,

["$S_", k, "$"])
od
] ;
\end{texgraph}

Figure 135

\begin{texgraph}[name=figure135, file]
Graph image = [
Marges(0, 0, 0, 0), ModelView(central), DistCam(20),
view(-6, 6, -6, 6), size(7.5),
theta :=-10*deg, phi :=60*deg,
P1 :=planEqn([1, 1, 1, 2]), P2 :=[Origin, vecK-vecJ],
D := interPP(P1, P2),
a :=Copy(getdroite(D), 1, 2),
b :=Copy(getplan(P1, 0.75), 11, 2),
c :=Copy(getplan(P2, 0.75), 3, 2),
Build3D(

bdPlan(P1, [color :=red, opacity :=0.7,
scale :=0.75 ]),

bdPlan(P2, [color :=blue, opacity :=0.7,
scale :=0.75]),

bdDroite(D, [color :=darkgreen,
width :=12]),

bdAxes([0, 0], [color :=gold,
width :=8, arrows :=1]),

bdLabel(a, "$D$", [labelpos :=[0.5, -i]]),
bdLabel(b, "$P_1$", [labelpos :=[0.5, i]]),
bdLabel(c, "$P_2$", [labelpos :=[0.5, i]])

),
Display3D()
] ;
\end{texgraph}

Figure 136

\begin{texgraph}[name=figure136, file]
Graph image = [
view(-5.5, 5.5, -5.5, 5.5), Marges(0, 0, 0, 0),
size(7.5), background(full, lightgray),
z :=-2,
Build3D(

bdPlan([M(0, 0, z), vecK],
[color :=gold, border :=1, bordercolor :=black]),

bdCylinder(M(-2, 3, 2), 7*M(2/3, -1, -2/3), 1,
[color :=slategray, smooth :=1]),

bdSphere(Origin, 3,
[color :=darkseagreen, clip :=-1,
clipwin :=[M(2, 1, 1), M(-1, -1, -1)],
smooth :=1, backculling :=0]),

bdCercle(M(0, 0, z), sqrt(5), vecK,
[color :=blue, width :=12]),

bdAxes([0, 0],
[arrows :=1, labels :=0, color :=firebrick])
),

Display3D(),
LineStyle :=dotted, Width :=6,
Color :=firebrick, Axes3D(0, 0, 0, 0)
] ;
\end{texgraph}

Figure 137

\begin{texgraph}[name=figure137, file]
Cmd [Fenetre(-6+5.5*i, 6-5.5*i, 0.625+0.625*i), Marges(0, 0, 0, 0), Border(0)] ;

[OriginalCoord(1), IdMatrix()] ;
[theta :=0.0872, phi :=1.1345, IdMatrix3D(), ModelView(ortho)] ;

Var
A = [-4.5*i, 4] ;
B = [-4.5*i, -1] ;
C = [0, -5] ;

Mac
plan = [ a :=%1, type :=%2, Arrows :=0,

LabelDot(Proj3D(%1), "$A$", "E", 1, 0.2),
Width :=8,
DrawPlan( [a, vecK], vecJ, 2, 2, type),
angleD( Proj3D(a+vecK), Proj3D(a), Proj3D(a-vecJ), 0.15),
Arrows :=1,
Ligne( Proj3D( [a, a+vecK]), 0),
LabelDot( Proj3D([a+vecK]), "$\vec{u}$", "N", 0)
] ;

Graph objet1 = [
Width :=8, Marges(0, 0, 0, 0), size(7.5),
plan(A, 1), plan( A+3*vecJ, 2), plan( A+6*vecJ, 3), plan( A+9*vecJ, 4),
plan(B, -1), plan( B+3*vecJ, -2), plan( B+6*vecJ, -3), plan( B+9*vecJ, -4),
plan(C),
Arrows :=0, LabelSize :=footnotesize,
Label(-4.5+2.7564*i, "type=$1$"),
Label(-1.2529+2.7564*i, "type=$2$"),
Label(1.5+2.7564*i, "type=$3$"),
Label(4.4824+2.7564*i, "type=$4$"),
Label(-4.7471-2.0032*i, "type=$-1$"),
Label(-1.5-2.0032*i, "type=$-2$"),
Label(1.5-2.0032*i, "type=$-3$"),
Label(4.2529-2.0032*i, "type=$-4$"),
Label(-0.2471-5.2532*i, "pas de type")
] ;

\end{texgraph}

75 / 86



TeXgraph Exemples de graphiques

A

−→u

L1 ·−→v
‖−→v ‖

L2 ·−→w
‖−→w ‖

M4

M1 M2

M3

Figure 138

\begin{texgraph}[name=figure138, file]
Graph image = [
view(-5, 5, -5, 5), Marges(0, 0, 0, 0), size(9), DotStyle :=cross,
A :=0, M4 :=-4+3*i, M1 :=-2-3*i, M2 :=4-3*i, M3 :=2+3*i,
LabelDot(A, "$A$", "NE", 1), Arrows :=1, Ligne([A, A+2*i], 0),
Width :=4, Arrows :=0, angleD(A+i, A, A-1, 0.25),
LabelStyle :=scriptsize, LabelDot(A+i, "$\vec{u}$", "O"),
Width :=8, Color :=red, Arrows :=1, Ligne([M1, M2], 0),
Width :=4, Color :=black, LabelStyle :=left,
LabelDot((M1+M2)/2, "$\dfrac{L_1\cdot\vec{v}}{\|\vec{v}\|}$", "S"),
Width :=8, Color :=red, Ligne([M2, M3], 0),
Width :=4, Color :=black, LabelStyle :=top,
LabelDot((M2+M3)/2, "$\dfrac{L2\cdot\vec{w}}{\|\vec{w}\|}$", "E"),
Arrows :=0,
LabelDot(M4, "$M4$", "NO", 1),
LabelDot(M1, "$M1$", "SO", 1),
LabelDot(M2, "$M2$", "SE", 1),
LabelDot(M3, "$M3$", "NE", 1),
Ligne([M3, M4, M1], 0)
] ;
\end{texgraph}

Figure 139

\begin{texgraph}[name=figure139, file]
Graph image = [
view(-2, 3.5, -3, 3.5), Marges(0, 0, 0, 0),
size(6), viewDir(55, 60),
L :=[M(0, 3, 0), M(-0.5, 2.5, 0),

M(-1, 3, 0), M(-1, 2, 0), M(0, 2, 0)],
P :=rotLine(L, [Origin, vecI], 1, 0, 2*pi, 35),
DrawFacet(P, [color :=coral])
] ;
\end{texgraph}

Figure 140

\begin{texgraph}[name=figure140, file]
Graph image = [
view(-5, 5, -6, 6), Marges(0, 0, 0, 0), size(6),
FillStyle :=full, FillColor :=lightpink,
Color :=blue, Width :=8,
DpqGoneReg(-2.5+2.5*i, -2.5+0.25*i, [7, 3]),
Label(-2.5+5*i, "[7,3]"),
DpqGoneReg(2.5+2.5*i, 2.5+0.25*i, [7, 2]),
Label(2.5+5*i, "[7,2]"),
DpqGoneReg(-2.5-2.5*i, -2.5-0.25*i, [7, 1]),
Label(-2.5-5*i, "[7,1]"),
DpqGoneReg(2.5-2.5*i, 2.5-0.25*i, [6, 2]),
Label(2.5-5*i, "[6,2]")
] ;
\end{texgraph}

Figure 141

\begin{texgraph}[name=figure141, file]
Graph image = [
view(-7, 7, -7, 7), Marges(0, 0, 0, 0),
size(6), FillStyle :=full,
FillColor :=lightblue,
Dsurface( M(u, v, cos(u)+sin(v)),

-5+5*i,
-5+5*i, 25+25*i)

] ;
\end{texgraph}

A B

Figure 142

\begin{texgraph}[name=figure142, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
A :=set("A", 0, [rotation :=30]),
B :=set("B", 0, [rotation :=-30]),
C := cap(A, B), Color :=red, FillStyle :=full,
FillOpacity :=0.5, FillColor :=pink,
Ligne(C, 0)
] ;
\end{texgraph}

Figure 143

\begin{texgraph}[name=figure143, file]
Graph image = [
ModelView(central), Marges(0, 0, 0, 0), size(6),
background(full, darkblue),
P :=Parallelep(M(-2.5, -2.5, -2.5), 5*vecI,

5*vecJ, 5*vecK), nb :=15,
Y :=[1, 1, 0], R :=[1, 0, 0],
Build3D(

for k from 1 to nb do
P :=HollowFacet(P, 0.25+i, aux),
C :=((nb-k+1)*R+(k-1)*Y)/nb,
bdFacet(P, [color :=RgbL(C)]),
P :=aux, sep3D

od),
Display3D()
] ;
\end{texgraph}

A B

Figure 144

\begin{texgraph}[name=figure144, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
A :=setB("A", 0, [rotation :=30]),
B :=setB("B", 0, [rotation :=-30]),
C := setminusB(A, B), Color :=red, FillStyle :=full,
drawSet(C, [FillOpacity :=0.5, FillColor :=pink])
] ;
\end{texgraph}

Figure 145

\begin{texgraph}[name=figure145, file]
Graph image = [
view(-1, 1, -1, 1), view3D(-2, 2, -2, 2, -2, 2),
Marges(0, 0, 0, 0), size(6),
P :=Tetra(Origin, vecI, vecJ, vecK),
P' :=ClipFacet(P, [M(0, 0, 0.5), -vecK]),
DrawFacet( P', [color :=pink])
] ;
\end{texgraph}

Figure 146

\begin{texgraph}[name=figure146, file]
Graph image = [
view(-2, 3, -2.5, 3), Marges(0, 0, 0, 0), size(6),
cube :=Parallelep(Origin, 3*vecI, 3*vecJ, 3*vecK),
plan :=[M(3, 0, 0), -vecI-vecK/2],
S :=Section(plan, cube), FillStyle :=full,
FillColor :=firebrick, HideWidth :=8,
DrawPoly(S, 4)
] ;
\end{texgraph}

Figure 147

\begin{texgraph}[name=figure147, file]
Graph image = [
view(-5, 5, -5, 5), Marges(0, 0, 0, 0),
size(6), Arrows :=1, Width :=4,
Axes(0, 1+i), Arrows :=0,
Width :=8, Color :=red,
Implicit( sin(x*y) )
] ;
\end{texgraph}

Figure 148

\begin{texgraph}[name=figure148, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
L :=for z in polyreg(0, 3.5, 6)

do [z, 0] od,
S :=lineTube(L, 1, 12, 1),
DrawFlatFacet(S, [color :=steelblue,

backculling :=1])
] ;
\end{texgraph}

Figure 149

\begin{texgraph}[name=figure149, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
C :=curve2Cylinder([4*cos(t)+2*i*sin(t), -3],

-pi, pi, [i, 6]),
DrawFlatFacet(C, [color :=violet])
] ;
\end{texgraph}

Figure 150

\begin{texgraph}[name=figure150, file]
Graph image = [
ModelView(central), view(-6, 6, -6, 6),
view3D(-5, 5, -4, 4, -4, 4),
Marges(0, 0, 0, 0), size(6),
L := for z in Get(Cartesienne(sin(x)), 0)

do [z, 0] od,
Arrows :=1, Axes3D(0, 0, 0),
Arrows :=0, Width :=8,
Color :=red, Ligne( Proj3D(L), 0)
] ;
\end{texgraph}

Figure 151

\begin{texgraph}[name=figure151, file]
Graph image = [
view(-8, 8, -8, 8), Marges(0, 0, 0, 0), size(6),
ModelView(central), DistCam(30),
Color :=darkgray,
Ligne3D([grille3d(x, -5, 1+2*i), grille3d(y, -5, 1+2*i),

grille3d(z, -5)], 0),
Color :=black,
S :=curveTube([3*exp(i*t), t/3], 0.5,

-3*pi, 3*pi, 100, 12),
Color :=black, Width :=1,
DrawFlatFacet(S,

[color :=steelblue, contrast :=0.5])
] ;
\end{texgraph}

Figure 152

\begin{texgraph}[name=figure152, file]
Graph image = [
view(-6.5, 6.5, -3, 5.5), Marges(0, 0, 0, 0), size(7.5),
view3D(-3, 3, -3, 3, -3, 3), ModelView(central),
Width :=8, Color :=blue, FillStyle :=full,
FillColor :=lightcyan, Cercle3D(Origin, 3, vecK),
Arrows :=1, LabelSize :=scriptsize,
Width :=4, Color :=black,
AxeX3D(axeOrigin :=M(-3, -3, 0), tickdir :=-vecJ,

xlabelstyle :=right, tickpos :=0, xlimits :=[-3, 3.5],
legendpos :=1, xlabelsep :=0.15),

AxeY3D(axeOrigin :=M(-3, -3, 0), tickdir :=-vecI,
tickpos :=0, ylimits :=[-3, 3.5]),

AxeZ3D(axeOrigin :=M(-3, -3, 0), zlimits :=[0, 3.5],
tickdir :=M(1, -1, 0), zgradlimits :=[1, 3]),

LineStyle :=dashed, Arrows :=0,
Ligne3D([M(0, -3, 0), Origin, M(-3, 0, 0)], 0),
LabelDot3D(Origin, "$O$", "E", 1)
] ;
\end{texgraph}

Figure 153

\begin{texgraph}[name=figure153, file]
Graph image = [
view(-10.5, 2.5, -1.5, 4.5), Marges(0, 0, 0, 0),
size(6, 0), Arrows :=1, Width :=4,
Axes(0, 1+i), Arrows :=0,
LabelAxe(y, 4.25*i, "$x$"),
LabelAxe(x, 2, "$t$", 2),
Width :=8, Color :=red, tMin :=-10, tMax :=2,
EquaDif(y+i*(t*x+y+sin(t)), 0, -1+i/2, 1),
Color :=black, LabelStyle :=stacked,
Label(-6+2*i,
"$x''-tx'-x=\sin(t)$\\
avec $x(0)=-1$ et $x'(0)=\frac12$")
] ;
\end{texgraph}

non épointé

épointé

Figure 154

\begin{texgraph}[name=figure154, file]
Graph image = [

view(-3.5, 4.5, -3, 3), Marges(0, 0, 0, 0), size(6),
P :=Parallelep(Origin, 3*vecI, 3*vecJ, 3*vecK),
P1 :=Chanfrein(P, 0.25, 0), P2 :=Chanfrein(P, 0.25, 1),
DrawFacet( shift3d(P1, -2*vecJ), [color :=steelblue],

shift3d(P2, 2*vecJ), [color :=steelblue]),
Label(-2-1.5*i, "non épointé", 1.8-2.6*i, "épointé")

] ;
\end{texgraph}

FF′

Figure 155

\begin{texgraph}[name=figure155, file]
Graph image = [
view(-5, 5, -5, 5), Marges(0, 0, 0, 0),
size(6),
F :=sqrt(7), F' :=-F, {foyers}
Width :=1, Color :=darkgray,
for t from -pi to pi step 0.1 do
M :=4*cos(t)+3*i*sin(t),
Vn :=(M-F)/abs(M-F)+(M-F')/abs(M-F'),
Droite(M, M+Vn), {normale à l'ellipse}

od,
Width :=8, Color :=red,
Ellipse(0, 4, 3),
LabelDot(F, "$F$", "S", 1),
LabelDot(F', "$F'$", "S", 1)
] ;
\end{texgraph}

Figure 156

\begin{texgraph}[name=figure156, file]
Graph image = [
view(-2, 2, -2, 2), Marges(0, 0, 0, 0), size(6),
StrListInit( nom, -1, "$A$", "O", i, "$B$", "N",

1, "$C$", "E", -i, "$D$", "S"),
for k from 1 to nom(0) step 3 do

LabelDot( Eval(@nom(k)), @nom(k+1), @nom(k+2), 1)
od,
StrListKill(nom)
] ;
\end{texgraph}

Figure 157

\begin{texgraph}[name=figure157, file]
Graph image = [
view(-4, 4, -4, 5), Marges(0, 0, 0, 0),
size(6), Width :=8,
A :=-3+4*i, B :=3+i, C :=3-3*i, D :=-3-3*i,
C1 :=4.5*i, C2 :=-2*i, C3 :=2-i, C4 :=-2,
FillStyle :=full, FillColor :=lightblue, Color :=red,
Bezier(A, C1, C2, B, jump, C, C3, C4, D, jump, A),
FillStyle :=none, DotStyle :=cross,
DotScale :=2, Color :=black,
LabelDot(A, "$A$", "N", 1),
LabelDot(B, "$B$", "E", 1),
LabelDot(C, "$C$", "SE", 1),
LabelDot(D, "$D$", "SO", 1),
LabelDot(C1, "$C_1$", "E", 1),
LabelDot(C2, "$C_2$", "SO", 1),
LabelDot(C3, "$C_3$", "N", 1),
LabelDot(C4, "$C_4$", "N", 1),
LineStyle :=userdash,
DashPattern :=[5, 2, 0.5, 2], Width :=6,
LineCap :=round,
Ligne([A, C1, C2, B, jump, C, C3, C4, D], 0)
] ;
\end{texgraph}

Figure 158

\begin{texgraph}[name=figure158, file]
Graph image = [
view(-2.25, 3.75, -2, 5), Marges(0, 0, 0, 0), size(6),
A :=0, B :=3+i, C :=2+4*i,
DotScale :=2, Width :=8,
Ligne([B, A, C], 0), Color :=red,
LabelDot(A, "$A$", "S", 1),
LabelDot(B, "$B$", "N", 1),
LabelDot(C, "$C$", "SE", 1),
Arrows :=1, Color :=blue,
Arc(B, A, C, 1, -1),
Arc(B, A, C, 2, 1)
] ;
\end{texgraph}

A B

Figure 159

\begin{texgraph}[name=figure159, file]
Graph image = [
Marges(0, 0, 0, 0), size(6),
A :=setB("A", 0, [rotation :=30]),
B :=setB("B", 0, [rotation :=-30]),
C := cupB(A, B), Color :=red, FillStyle :=full,
drawSet(C, [FillOpacity :=0.5, FillColor :=pink])
] ;
\end{texgraph}

